
Abstract. The report gives a defining description of the programming language Scheme. Scheme is a statically
scoped and properly tail-recursive dialect of the Lisp programming language invented by Guy Lewis Steele, Jr. and
Gerald Jay Sussman. It was designed to have an exceptionally clear and simple semantics and few different ways
to form expressions. A wide variety of programming paradigms, including imperative, functional, and message
passing styles, find convenient expression in Scheme.

The introduction offers a brief history of the language and of the report.
The first three chapters present the fundamental ideas of the language and describe the notational conventions

used for describing the language and for writing programs in the language.
Sections 5 and 6 describe the syntax and semantics of expressions, programs, and definitions.
Section 7 describes Scheme’s built-in procedures, which include all of the language’s data manipulation and

input/output primitives.
Section 8 provides a formal syntax for Scheme written in extended BNF, along with a formal denotational

semantics. An example of the use of the language follows the formal syntax and semantics.
The report concludes with a list of references and an alphabetic index and is followed by a short list of

clarifications and corrections.

Keywords: Scheme, Lisp, programming languages

1. Introduction

Programming languages should be designed not by piling feature on top of feature, but by
removing the weaknesses and restrictions that make additional features appear necessary.
Scheme demonstrates that a very small number of rules for forming expressions, with no
restrictions on how they are composed, suffice to form a practical and efficient programming
language that is flexible enough to support most of the major programming paradigms in
use today.

Scheme was one of the first programming languages to incorporate first class procedures
as in the lambda calculus, thereby proving the usefulness of static scope rules and block
structure in a dynamically typed language. Scheme was the first major dialect of Lisp
to distinguish procedures from lambda expressions and symbols, to use a single lexical
environment for all variables, and to evaluate the operator position of a procedure call in the
same way as an operand position. By relying entirely on procedure calls to express iteration,

Scheme emphasized the fact that tail-recursive procedure calls are essentially goto’s that pass
arguments. Scheme was the first widely used programming language to embrace first class
escape procedures, from which all previously known sequential control structures can be
synthesized. A subsequent version of Scheme introduced the concept of exact and inexact
numbers, an extension of Common Lisp’s generic arithmetic. More recently, Scheme
became the first programming language to support hygienic macros, which permit the
syntax of a block-structured language to be extended in a consistent and reliable manner.

Background. The first description of Scheme was written in 1975 [28]. A revised report
[25] appeared in 1978, which described the evolution of the language as its MIT implemen-
tation was upgraded to support an innovative compiler [26]. Three distinct projects began
in 1981 and 1982 to use variants of Scheme for courses at MIT, Yale, and Indiana University
[10, 17, 21]. An introductory computer science textbook using Scheme was published in
1984 [1].

As Scheme became more widespread, local dialects began to diverge until students and
researchers occasionally found it difficult to understand code written at other sites. Fifteen
representatives of the major implementations of Scheme therefore met in October 1984 to
work toward a better and more widely accepted standard for Scheme. Their report [4] was
published at MIT and Indiana University in the summer of 1985. Further revision took
place in the spring of 1986 [23], and in the spring of 1988 [6]. The present report reflects
further revisions agreed upon in a meeting at Xerox PARC in June 1992.

We intend this report to belong to the entire Scheme community, and so we grant permis-
sion to copy it in whole or in part without fee. In particular, we encourage implementors of
Scheme to use this report as a starting point for manuals and other documentation, modifying
it as necessary.

2. Overview of Scheme

2.1. Semantics

This section gives an overview of Scheme’s semantics. A detailed informal semantics is
the subject of Sections 4 through 7. For reference purposes, Section 8.2 provides a formal
semantics of Scheme.

Following Algol, Scheme is a statically scoped programming language. Each use of a
variable is associated with a lexically apparent binding of that variable.

Scheme has latent as opposed to manifest types. Types are associated with values (also
called objects) rather than with variables. (Some authors refer to languages with latent
types as weakly typed or dynamically typed languages.) Other languages with latent types
are APL, Snobol, and other dialects of Lisp. Languages with manifest types (sometimes
referred to as strongly typed or statically typed languages) include Algol 60, Pascal, and C.

All objects created in the course of a Scheme computation, including procedures and
continuations, have unlimited extent. No Scheme object is ever destroyed. The reason that
implementations of Scheme do not (usually!) run out of storage is that they are permitted to
reclaim the storage occupied by an object if they can prove that the object cannot possibly

matter to any future computation. Other languages in which most objects have unlimited
extent include APL and other Lisp dialects.

Implementations of Scheme are required to be properly tail-recursive. This allows the
execution of an iterative computation in constant space, even if the iterative computation
is described by a syntactically recursive procedure. Thus with a properly tail-recursive
implementation, iteration can be expressed using the ordinary procedure-call mechanics,
so that special iteration constructs are useful only as syntactic sugar. See Section 4.5.

Scheme procedures are objects in their own right. Procedures can be created dynamically,
stored in data structures, returned as results of procedures, and so on. Other languages with
these properties include Common Lisp and ML.

One distinguishing feature of Scheme is that continuations, which in most other languages
only operate behind the scenes, also have “first-class” status. Continuations are useful
for implementing a wide variety of advanced control constructs, including nonlocal exits,
backtracking, and coroutines. See Section 7.4.

Arguments to Scheme procedures are always passed by value, which means that the
actual argument expressions are evaluated before the procedure gains control, whether the
procedure needs the result of the evaluation or not. ML, C, and APL are three other languages
that always pass arguments by value. This is distinct from the lazy-evaluation semantics of
Haskell, or the call-by-name semantics of Algol 60, where an argument expression is not
evaluated unless its value is needed by the procedure.

Scheme’s model of arithmetic is designed to remain as independent as possible of the
particular ways in which numbers are represented within a computer. In Scheme, every
integer is a rational number, every rational is a real, and every real is a complex number.
Thus the distinction between integer and real arithmetic, so important to many programming
languages, does not appear in Scheme. In its place is a distinction between exact arithmetic,
which corresponds to the mathematical ideal, and inexact arithmetic on approximations.
As in Common Lisp, exact arithmetic is not limited to integers.

2.2. Syntax

Scheme, like most dialects of Lisp, employs a fully parenthesized prefix notation for pro-
grams and (other) data; the grammar of Scheme generates a sublanguage of the language
used for data. An important consequence of this simple, uniform representation is the sus-
ceptibility of Scheme programs and data to uniform treatment by other Scheme programs.
For example, theeval procedure evaluates a Scheme program expressed as data.

Theread procedure performs syntactic as well as lexical decomposition of the data it
reads. Theread procedure parses its input as data (Section 8.1.2), not as program.

The formal syntax of Scheme is described in Section 8.1.

2.3. Notation and terminology

2.3.1. Primitive, library, and optional features.It is required that every implementation
of Scheme support all features that are not marked as beingoptional. Implementations are
free to omit optional features of Scheme or to add extensions, provided the extensions are

not in conflict with the language reported here. In particular, implementations must support
portable code by providing a syntactic mode that preempts no lexical conventions of this
report.

To aid in understanding and implementing Scheme, some features are marked aslibrary.
These can be easily implemented in terms of the other, primitive, features. They are
redundant in the strict sense of the word, but they capture common patterns of usage, and
are therefore provided as convenient abbreviations.

2.3.2. Error situations and unspecified behavior.When speaking of an error situation, this
report uses the phrase “an error is signalled” to indicate that implementations must detect
and report the error. If such wording does not appear in the discussion of an error, then
implementations are not required to detect or report the error, though they are encouraged to
do so. An error situation that implementations are not required to detect is usually referred
to simply as “an error.”

For example, it is an error for a procedure to be passed an argument that the procedure
is not explicitly specified to handle, even though such domain errors are seldom mentioned
in this report. Implementations may extend a procedure’s domain of definition to include
such arguments.

This report uses the phrase “may report a violation of an implementation restriction” to
indicate circumstances under which an implementation is permitted to report that it is unable
to continue execution of a correct program because of some restriction imposed by the im-
plementation. Implementation restrictions are of course discouraged, but implementations
are encouraged to report violations of implementation restrictions.

For example, an implementation may report a violation of an implementation restriction
if it does not have enough storage to run a program.

If the value of an expression is said to be “unspecified,” then the expression must evaluate
to some object without signalling an error, but the value depends on the implementation;
this report explicitly does not say what value should be returned.

2.3.3. Entry format. Sections 5 and 7 are organized into entries. Each entry describes
one language feature or a group of related features, where a feature is either a syntactic
construct or a built-in procedure. An entry begins with one or more header lines of the form

template category

for required, primitive features, or

template qualifier category

wherequalifier is either “library” or “optional” as defined in Section 2.3.1.
If categoryis “syntax”, the entry describes an expression type, and the template gives

the syntax of the expression type. Components of expressions are designated by syntactic
variables, which are written using angle brackets, for example,〈expression〉, 〈variable〉.
Syntactic variables should be understood to denote segments of program text; for example,

〈expression〉 stands for any string of characters which is a syntactically valid expression.
The notation

〈thing1〉 . . .

indicates zero or more occurrences of a〈thing〉, and

〈thing1〉 〈thing2〉 . . .

indicates one or more occurrences of a〈thing〉.
If categoryis “procedure”, then the entry describes a procedure, and the header line gives

a template for a call to the procedure. Argument names in the template areitalicized. Thus
the header line

(vector-ref vector k) procedure

indicates that the built-in procedurevector-ref takes two arguments, a vectorvectorand
an exact non-negative integerk (see below). The header lines

(make-vector k) procedure
(make-vector k fill) procedure

indicate that themake-vector procedure must be defined to take either one or two argu-
ments.

It is an error for an operation to be presented with an argument that it is not specified to
handle. For succinctness, we follow the convention that if an argument name is also the name
of a type listed in Section 4.2, then that argument must be of the named type. For example,
the header line forvector-ref given above dictates that the first argument tovector-ref
must be a vector. The following naming conventions also imply type restrictions:

obj any object
list, list1, . . . , list j , . . . list (see Section 7.3.2)
z, z1, . . . zj , . . . complex number
x, x1, . . . xj , . . . real number
y, y1, . . . yj , . . . real number
q,q1, . . .qj , . . . rational number
n, n1, . . .nj , . . . integer
k, k1, . . . kj , . . . exact nonnegative integer

2.3.4. Evaluation examples.The symbol “H⇒” used in program examples should be read
“evaluates to.” For example,

(* 5 8) H⇒ 40

means that the expression(* 5 8) evaluates to the object40. Or, more precisely: the
expression given by the sequence of characters “(* 5 8)” evaluates, in the initial environ-
ment, to an object that may be represented externally by the sequence of characters “40”.
See Section 4.3 for a discussion of external representations of objects.

2.3.5. Naming conventions.By convention, the names of procedures that always return a
boolean value usually end in “?”. Such procedures are called predicates.

By convention, the names of procedures that store values into previously allocated loca-
tions (see Section 4.4) usually end in “!”. Such procedures are called mutation procedures.
By convention, the value returned by a mutation procedure is unspecified.

By convention, “->” appears within the names of procedures that take an object of one
type and return an analogous object of another type. For example,list->vector takes a
list and returns a vector whose elements are the same as those of the list.

3. Lexical conventions

This section gives an informal account of some of the lexical conventions used in writing
Scheme programs. For a formal syntax of Scheme, see Section 8.1.

Upper and lower case forms of a letter are never distinguished except within character
and string constants. For example, Foo is the same identifier asFOO, and#x1AB is the same
number as#X1ab.

3.1. Identifiers

Most identifiers allowed by other programming languages are also acceptable to Scheme.
The precise rules for forming identifiers vary among implementations of Scheme, but in
all implementations a sequence of letters, digits, and “extended alphabetic characters” that
begins with a character that cannot begin a number is an identifier. In addition,+, -, and
... are identifiers. Here are some examples of identifiers:

lambda q
list->vector soup
+ V17a
<=? a34kTMNs
the-word-recursion-has-many-meanings

Extended alphabetic characters may be used within identifiers as if they were letters. The
following are extended alphabetic characters:

! $ % & * + - . / : < = > ? @ ^ _ ~

See Section 8.11 for a formal syntax of identifiers.
Identifiers have two uses within Scheme programs:

• Any identifier may be used as a variable or as a syntactic keyword (see Sections 4.1
and 5.3).

• When an identifier appears as a literal or within a literal (see Section 5.1.2), it is being
used to denote asymbol(see Section 7.3.3).

3.2. Whitespace and comments

Whitespacecharacters are spaces and newlines. (Implementations typically provide addi-
tional whitespace characters such as tab or page break.) Whitespace is used for improved
readability and as necessary to separate tokens from each other, a token being an indivisible
lexical unit such as an identifier or number, but is otherwise insignificant. Whitespace may
occur between any two tokens, but not within a token. Whitespace may also occur inside a
string, where it is significant.

A semicolon (;) indicates the start of a comment. The comment continues to the end of
the line on which the semicolon appears. Comments are invisible to Scheme, but the end
of the line is visible as whitespace. This prevents a comment from appearing in the middle
of an identifier or number.

;;; The FACT procedure computes the factorial
;;; of a nonnegative integer.
(define fact
(lambda (n)
(if (= n 0)

1 ;Base case: return 1
(* n (fact (-n 1))))))

3.3. Other notations

For a description of the notations used for numbers, see Section 7.2.

. + - These are used in numbers, and may also occur anywhere in an identifier except
as the first character. A delimited plus or minus sign by itself is also an identifier. A
delimited period (not occurring within a number or identifier) is used in the notation
for pairs (Section 7.3.2), and to indicate a rest-parameter in a formal parameter list
(Section 5.1.4). A delimited sequence of three successive periods is also an identifier.

( ) Parentheses are used for grouping and to notate lists (Section 7.3.2).
’ The single quote character is used to indicate literal data (Section 5.1.2).
` The backquote character is used to indicate almost-constant data (Section 5.2.6).
, ,@ The character comma and the sequence comma at-sign are used in conjunction with

backquote (Section 5.2.6).
" The double quote character is used to delimit strings (Section 7.3.5).
\ Backslash is used in the syntax for character constants (Section 7.3.4) and as an escape

character within string constants (Section 7.3.5).
[ ] { } | Left and right square brackets and curly braces and vertical bar are reserved

for possible future extensions to the language.

# Sharp sign is used for a variety of purposes depending on the character that immediately
follows it:

#t #f These are the boolean constants (Section 7.3.1).
#\ This introduces a character constant (Section 7.3.4).
#( This introduces a vector constant (Section 7.3.6). Vector constants are terminated

by ).
#e #i # b #o #d #x These are used in the notation for numbers (Section 7.2.4).

4. Basic concepts

4.1. Variables, syntactic keywords, and regions

An identifier may name a type of syntax, or it may name a location where a value can be
stored. An identifier that names a type of syntax is called asyntactic keywordand is said
to beboundto that syntax. An identifier that names a location is called avariableand is
said to beboundto that location. The set of all visible bindings in effect at some point in a
program is known as theenvironmentin effect at that point. The value stored in the location
to which a variable is bound is called the variable’s value. By abuse of terminology, the
variable is sometimes said to name the value or to be bound to the value. This is not quite
accurate, but confusion rarely results from this practice.

Certain expression types are used to create new kinds of syntax and bind syntactic key-
words to those new syntaxes, while other expression types create new locations and bind
variables to those locations. These expression types are calledbinding constructs. Those
that bind syntactic keywords are listed in Section 5.3. The most fundamental of the variable
binding constructs is thelambda expression, because all other variable binding constructs
can be explained in terms oflambda expressions. The other variable binding constructs
arelet, let*, letrec, anddo expressions (see Sections 5.1.4, 5.2.2, and 5.2.4).

Like Algol and Pascal, and unlike most other dialects of Lisp except for Common Lisp,
Scheme is a statically scoped language with block structure. To each place where an
identifier is bound in a program there corresponds aregionof the program text within which
the binding is visible. The region is determined by the particular binding construct that
establishes the binding; if the binding is established by alambda expression, for example,
then its region is the entirelambda expression. Every mention of an identifier refers to the
binding of the identifier that established the innermost of the regions containing the use. If
there is no binding of the identifier whose region contains the use, then the use refers to the
binding for the variable in the top level environment, if any (Sections 5 and 7); if there is
no binding for the identifier, it is said to beunbound.

4.2. Disjointness of types

No object satisfies more than one of the following predicates:

boolean? pair?
symbol? number?

char? string?
vector? port?
procedure?

These predicates define the typesboolean, pair, symbol, number, char (or character),
string, vector, port, andprocedure. The empty list is a special object of its own type; it
satisfies none of the above predicates.

Although there is a separate boolean type, any Scheme value can be used as a boolean
value for the purpose of a conditional test. As explained in Section 7.3.1, all values count
as true in such a test except for#f. This report uses the word “true” to refer to any Scheme
value except#f, and the word “false” to refer to#f.

4.3. External representations

An important concept in Scheme (and Lisp) is that of theexternal representationof an
object as a sequence of characters. For example, an external representation of the integer
28 is the sequence of characters “28”, and an external representation of a list consisting of
the integers 8 and 13 is the sequence of characters “(8 13)”.

The external representation of an object is not necessarily unique. The integer 28 also
has representations “#e28.000” and “#x1c”, and the list in the previous paragraph also
has the representations “( 08 13 )” and “( 8 . (13.()))” (see Section 7.3.2).

Many objects have standard external representations, but some, such as procedures, do
not have standard representations (although particular implementations may define repre-
sentations for them).

An external representation may be written in a program to obtain the corresponding object
(seequote, Section 5.1.2).

External representations can also be used for input and output. The procedureread
(Section 7.6.2) parses external representations, and the procedurewrite (Section 7.6.3)
generates them. Together, they provide an elegant and powerful input/output facility.

Note that the sequence of characters “(+ 2 6)” is not an external representation of the
integer 8, even though itis an expression evaluating to the integer 8; rather, it is an external
representation of a three-element list, the elements of which are the symbol+ and the in-
tegers 2 and 6. Scheme’s syntax has the property that any sequence of characters that is an
expression is also the external representation of some object. This can lead to confusion,
since it may not be obvious out of context whether a given sequence of characters is in-
tended to denote data or program, but it is also a source of power, since it facilitates writing
programs such as interpreters and compilers that treat programs as data (or vice versa).

The syntax of external representations of various kinds of objects accompanies the de-
scription of the primitives for manipulating the objects in the appropriate subsections of
Section 7.

4.4. Storage model

Variables and objects such as pairs, vectors, and strings implicitly denote locations or
sequences of locations. A string, for example, denotes as many locations as there are

characters in the string. (These locations need not correspond to a full machine word.) A
new value may be stored into one of these locations using thestring-set! procedure, but
the string continues to denote the same locations as before.

An object fetched from a location, by a variable reference or by a procedure such ascar,
vector-ref, orstring-ref, is equivalent in the sense ofeqv? (Section 7.1) to the object
last stored in the location before the fetch.

Every location is marked to show whether it is in use. No variable or object ever refers
to a location that is not in use. Whenever this report speaks of storage being allocated for
a variable or object, what is meant is that an appropriate number of locations are chosen
from the set of locations that are not in use, and the chosen locations are marked to indicate
that they are now in use before the variable or object is made to denote them.

In many systems it is desirable for constants (i.e., the values of literal expressions) to
reside in read-only-memory. To express this, it is convenient to imagine that every object
that denotes locations is associated with a flag telling whether that object is mutable or
immutable. In such systems literal constants and the strings returned bysymbol->string
are immutable objects, while all objects created by the other procedures listed in this report
are mutable. It is an error to attempt to store a new value into a location that is denoted by
an immutable object.

4.5. Proper tail recursion

Implementations of Scheme are required to beproperly tail-recursive. Procedure calls that
occur in certain syntactic contexts defined below are ‘tail calls’. A Scheme implementation
is properly tail-recursive if it supports an unbounded number of active tail calls. A call is
active if the called procedure may still return. Note that this includes calls that may be
returned from either by the current continuation or by continuations captured earlier by
call-with-current-continuation that are later invoked. In the absence of captured
continuations, calls could return at most once and the active calls would be those that had
not yet returned. A formal definition of proper tail recursion can be found in [8].

Rationale:Intuitively, no space is needed for an active tail call because the continuation that
is used in the tail call has the same semantics as the continuation passed to the procedure
containing the call. Although an improper implementation might use a new continuation
in the call, a return to this new continuation would be followed immediately by a return to
the continuation passed to the procedure. A properly tail-recursive implementation returns
to that continuation directly.

Proper tail recursion was one of the central ideas in Steele and Sussman’s original version
of Scheme. Their first Scheme interpreter implemented both functions and actors. Control
flow was expressed using actors, which differed from functions in that they passed their
results on to another actor instead of returning to a caller. In the terminology of this section,
each actor finished with a tail call to another actor.

Steele and Sussman later observed that in their interpreter the code for dealing with
actors was identical to that for functions and thus there was no need to include both in the
language.

A tail call is a procedure call that occurs in atail context. Tail contexts are defined
inductively. Note that a tail context is always determined with respect to a particular lambda
expression.

• The last expression within the body of a lambda expression, shown as〈tail expression〉
below, occurs in a tail context.

( lambda 〈formals〉
〈definition〉∗ 〈expression〉∗ 〈tail expression〉)

• If one of the following expressions is in a tail context, then the subexpressions shown
as〈tail expression〉 are in a tail context. These were derived from rules in the grammar
given in Section 8 by replacing some occurrences of〈expression〉 with 〈tail expression〉.
Only those rules that contain tail contexts are shown here.

(if 〈expression〉 〈tail expression〉 〈tail expression〉)
(if 〈expression〉 〈tail expression〉)

(cond 〈cond clause〉+)
(cond 〈cond clause〉∗ (else 〈tail sequence〉))
(case 〈expression〉
〈case clause〉+)

(case 〈expression〉
〈case clause〉∗
( else 〈tail sequence〉))

(and 〈expression〉∗ 〈tail expression〉)
(or 〈expression〉∗ 〈tail expression〉)
(let (〈binding spec〉∗) 〈tail body〉)
(let 〈variable〉 (〈binding spec〉∗) 〈tail body〉)
(let∗ (〈binding spec〉∗) 〈tail body〉)
(letrec (〈binding spec〉∗) 〈tail body〉)
(let-syntax (〈syntax spec〉∗) 〈tail body〉)
(letrec-syntax (〈syntax spec〉∗) 〈tail body〉)
(begin 〈tail sequence〉)
(do (〈iteration spec〉∗)

(〈test〉 〈tail sequence〉)
〈expression〉∗)

where

〈cond clause〉 −→( 〈test〉 〈tail sequence〉)
〈case clause〉 −→((〈datum〉∗) 〈tail sequence〉)

〈tail body〉 −→ 〈definition〉∗ 〈tail sequence〉
〈tail sequence〉 −→ 〈expression〉∗ 〈tail expression〉

• If a cond expression is in a tail context, and has a clause of the form(〈expression1〉 =>
〈expression2〉) then the (implied) call to the procedure that results from the evaluation of
〈expression2〉 is in a tail context.〈expression2〉 itself is not in a tail context.

Certain built-in procedures are also required to perform tail calls. The first argument
passed toapply and tocall-with-current-continuation, and the second argument
passed tocall-with-values, must be called via a tail call. Similarly,evalmust evaluate
its argument as if it were in tail position within theeval procedure.

In the following example the only tail call is the call tof. None of the calls tog or h are
tail calls. The reference tox is in a tail context, but it is not a call and thus is not a tail call.

( lambda ()
(if (g)

(let ((x (h)))
x)

(and (g) (f))))

Note: Implementations are allowed, but not required, to recognize that some nontail calls,
such as the call toh above, can be evaluated as though they were tail calls. In the example
above, thelet expression could be compiled as a tail call toh. (The possibility ofh
returning an unexpected number of values can be ignored, because in that case the effect of
thelet is explicitly unspecified and implementation-dependent.)

5. Expressions

Expression types are categorized asprimitiveorderived. Primitive expression types include
variables and procedure calls. Derived expression types are not semantically primitive,
but can instead be defined as macros. With the exception ofquasiquote, whose macro
definition is complex, the derived expressions are classified as library features. Suitable
definitions are given in Section 8.3.

5.1. Primitive expression types

5.1.1. Variable references.

〈variable〉 syntax

An expression consisting of a variable (Section 4.1) is a variable reference. The value of
the variable reference is the value stored in the location to which the variable is bound. It
is an error to reference an unbound variable.

(define x 28)
x H⇒ 28

5.1.2. Literal expressions.

(quote 〈datum〉) syntax
’〈datum〉 syntax
〈constant〉 syntax

(quote 〈datum〉) evaluates to〈datum〉. 〈Datum〉 may be any external representation
of a Scheme object (see Section 4.3). This notation is used to include literal constants in
Scheme code.

(quote a) H⇒ a
(quote #(a b c)) H⇒ #(a b c)
(quote (+ 1 2)) H⇒ (+ 1 2)

(quote 〈datum〉) may be abbreviated as’〈datum〉. The two notations are equivalent in
all respects.

’a H⇒ a
’#(a b c) H⇒ #(a b c)
’() H⇒ ()
’(+ 1 2) H⇒ (+ 1 2)
’(quote a) H⇒ (quote a)
’’a H⇒ (quote a)

Numerical constants, string constants, character constants, and boolean constants evalu-
ate “to themselves”; they need not be quoted.

’"abc" H⇒ "abc"
"abc" H⇒ "abc"
’145932 H⇒ 145932
145932 H⇒ 145932
’#t H⇒ #t
#t H⇒ #t

As noted in Section 4.4, it is an error to alter a constant (i.e., the value of a literal
expression) using a mutation procedure likeset-car! or string-set!.

5.1.3. Procedure calls.

(〈operator〉 〈operand1〉 . . .) syntax

A procedure call is written by simply enclosing in parentheses expressions for the proce-
dure to be called and the arguments to be passed to it. The operator and operand expressions
are evaluated (in an unspecified order) and the resulting procedure is passed the resulting
arguments.

(+ 3 4) H⇒ 7
((if #f + *) 3 4) H⇒ 12

A number of procedures are available as the values of variables in the initial environment;
for example, the addition and multiplication procedures in the above examples are the values
of the variables+ and*. New procedures are created by evaluatinglambda expressions
(see Section 5.1.4).

Procedure calls may return any number of values (seevalues in Section 7.4). With the
exception ofvalues the procedures available in the initial environment return one value
or, for procedures such asapply, pass on the values returned by a call to one of their
arguments.

Procedure calls are also calledcombinations.

Note: In contrast to other dialects of Lisp, the order of evaluation is unspecified, and
the operator expression and the operand expressions are always evaluated with the same
evaluation rules.

Note: Although the order of evaluation is otherwise unspecified, the effect of any con-
current evaluation of the operator and operand expressions is constrained to be consistent
with some sequential order of evaluation. The order of evaluation may be chosen differently
for each procedure call.

Note: In many dialects of Lisp, the empty combination,(), is a legitimate expression.
In Scheme, combinations must have at least one subexpression, so() is not a syntactically
valid expression.

5.1.4. Procedures.

(lambda 〈formals〉 〈body〉) syntax

Syntax:〈Formals〉 should be a formal arguments list as described below, and〈body〉 should
be a sequence of one or more expressions.

Semantics:A lambda expression evaluates to a procedure. The environment in effect
when thelambda expression was evaluated is remembered as part of the procedure. When
the procedure is later called with some actual arguments, the environment in which the
lambda expression was evaluated will be extended by binding the variables in the formal
argument list to fresh locations, the corresponding actual argument values will be stored in
those locations, and the expressions in the body of thelambda expression will be evaluated

sequentially in the extended environment. The result(s) of the last expression in the body
will be returned as the result(s) of the procedure call.

(lambda (x) (+ x x)) H⇒ a procedure
((lambda (x) (+ x x)) 4) H⇒ 8

(define reverse-subtract
(lambda (x y) (- y x)))

(reverse-subtract 7 10) H⇒ 3

(define add4
(let ((x 4))

(lambda (y) (+ x y))))
(add4 6) H⇒ 10

〈Formals〉 should have one of the following forms:

• (〈variable1〉 . . .): The procedure takes a fixed number of arguments; when the procedure
is called, the arguments will be stored in the bindings of the corresponding variables.
• 〈variable〉: The procedure takes any number of arguments; when the procedure is called,

the sequence of actual arguments is converted into a newly allocated list, and the list is
stored in the binding of the〈variable〉.
• (〈variable1〉 . . . 〈variablen〉 . 〈variablen+1〉): If a space-delimited period precedes the last

variable, then the procedure takesn or more arguments, wheren is the number of formal
arguments before the period (there must be at least one). The value stored in the binding
of the last variable will be a newly allocated list of the actual arguments left over after all
the other actual arguments have been matched up against the other formal arguments.

It is an error for a〈variable〉 to appear more than once in〈formals〉.

((lambda x x) 3 4 5 6) H⇒ (3 4 5 6)
((lambda (x y . z) z)
3 4 5 6) H⇒ (5 6)

Each procedure created as the result of evaluating alambda expression is (conceptually)
tagged with a storage location, in order to makeeqv? andeq? work on procedures (see
Section 7.1).

5.1.5. Conditionals.

(if 〈test〉 〈consequent〉 〈alternate〉) syntax
(if 〈test〉 〈consequent〉) syntax

Syntax:〈Test〉, 〈consequent〉, and〈alternate〉 may be arbitrary expressions.

Semantics:An if expression is evaluated as follows: first,〈test〉 is evaluated. If it yields
a true value (see Section 7.3.1), then〈consequent〉 is evaluated and its value(s) is(are) re-
turned. Otherwise〈alternate〉 is evaluated and its value(s) is(are) returned. If〈test〉 yields
a false value and no〈alternate〉 is specified, then the result of the expression is unspecified.

(if (> 3 2) ’yes ’no) H⇒ yes
(if (> 2 3) ’yes ’no) H⇒ no
(if (> 3 2)

(- 3 2)
(+ 3 2)) H⇒ 1

5.1.6. Assignments.

(set! 〈variable〉 〈expression〉) syntax

〈Expression〉 is evaluated, and the resulting value is stored in the location to which〈variable〉
is bound. 〈Variable〉 must be bound either in some region enclosing theset! expression
or at top level. The result of theset! expression is unspecified.

(define x 2)
(+ x 1) H⇒ 3
(set! x 4) H⇒ unspecified
(+ x 1) H⇒ 5

5.2. Derived expression types

The constructs in this section are hygienic, as discussed in Section 5.3. For reference pur-
poses, Section 8.3 gives macro definitions that will convert most of the constructs described
in this section into the primitive constructs described in the previous section.

5.2.1. Conditionals.

(cond 〈clause1〉 〈clause2〉 . . .) library syntax

Syntax:Each〈clause〉 should be of the form

(〈test〉 〈expression1〉 . . .)

where〈test〉 is any expression. Alternatively, a〈clause〉 may be of the form

(〈test〉 => 〈expression〉)

The last〈clause〉 may be an “else clause,” which has the form

(else 〈expression1〉 〈expression2〉 . . .).

Semantics:A condexpression is evaluated by evaluating the〈test〉expressions of successive
〈clause〉s in order until one of them evaluates to a true value (see Section 7.3.1). When a
〈test〉 evaluates to a true value, then the remaining〈expression〉s in its〈clause〉 are evaluated
in order, and the result(s) of the last〈expression〉 in the 〈clause〉 is(are) returned as the
result(s) of the entirecond expression. If the selected〈clause〉 contains only the〈test〉
and no〈expression〉s, then the value of the〈test〉 is returned as the result. If the selected
〈clause〉 uses the=> alternate form, then the〈expression〉 is evaluated. Its value must be a
procedure that accepts one argument; this procedure is then called on the value of the〈test〉
and the value(s) returned by this procedure is(are) returned by thecond expression. If all
〈test〉s evaluate to false values, and there is no else clause, then the result of the conditional
expression is unspecified; if there is an else clause, then its〈expression〉s are evaluated, and
the value(s) of the last one is(are) returned.

(cond ((> 3 2) ’greater)
((< 3 2) ’less)) H⇒ greater

(cond ((> 3 3) ’greater)
((< 3 3) ’less)
(else ’equal)) H⇒ equal

(cond ((assv ’b ’((a 1)(b 2))) => cadr)
(else #f)) H⇒ 2

(case 〈key〉 〈clause1〉 〈clause2〉 . . .) library syntax

Syntax:〈Key〉 may be any expression. Each〈clause〉 should have the form

((〈datum1〉 . . .) 〈expression1〉 〈expression2〉 . . . ),

where each〈datum〉 is an external representation of some object. All the〈datum〉s must be
distinct. The last〈clause〉 may be an “else clause,” which has the form

(else 〈expression1〉 〈expression2〉 . . . ).

Semantics:A case expression is evaluated as follows.〈Key〉 is evaluated and its result is
compared against each〈datum〉. If the result of evaluating〈key〉 is equivalent (in the sense
of eqv?; see Section 7.1) to a〈datum〉, then the expressions in the corresponding〈clause〉
are evaluated from left to right and the result(s) of the last expression in the〈clause〉 is(are)
returned as the result(s) of thecase expression. If the result of evaluating〈key〉 is different
from every〈datum〉, then if there is an else clause its expressions are evaluated and the
result(s) of the last is(are) the result(s) of thecase expression; otherwise the result of the
case expression is unspecified.

(case (* 2 3)
((2 3 5 7) ’prime)
((1 4 6 8 9) ’composite)) H⇒ composite

(case (car ’(c d))
((a) ’a)
((b) ’b)) H⇒ unspecified

(case (car ’(c d))
((a e i o u)’vowel)
((w y) ’semivowel)
(else ’consonant)) H⇒ consonant

(and 〈test1〉 . . .) library syntax

The〈test〉 expressions are evaluated from left to right, and the value of the first expression
that evaluates to a false value (see Section 7.3.1) is returned. Any remaining expressions are
not evaluated. If all the expressions evaluate to true values, the value of the last expression
is returned. If there are no expressions then#t is returned.

(and (= 2 2)(> 2 1)) H⇒ #t
(and (= 2 2)(< 2 1)) H⇒ #f
(and 1 2 ’c ’(f g)) H⇒ (f g)
(and) H⇒ #t

(or 〈test1〉 . . .) library syntax

The〈test〉 expressions are evaluated from left to right, and the value of the first expression
that evaluates to a true value (see Section 7.3.1) is returned. Any remaining expressions are
not evaluated. If all expressions evaluate to false values, the value of the last expression is
returned. If there are no expressions then#f is returned.

(or (= 2 2) (> 2 1)) H⇒ #t
(or (= 2 2)( < 2 1)) H⇒ #t
(or #f #f #f) H⇒ #f
(or (memq ’b ’(a b c))

(/ 3 0)) H⇒ (b c)

5.2.2. Binding constructs. The three binding constructslet, let*, andletrec give
Scheme a block structure, like Algol 60. The syntax of the three constructs is identical, but
they differ in the regions they establish for their variable bindings. In alet expression, the
initial values are computed before any of the variables become bound; in alet* expression,
the bindings and evaluations are performed sequentially; while in aletrec expression, all
the bindings are in effect while their initial values are being computed, thus allowing
mutually recursive definitions.

(let 〈bindings〉 〈body〉) library syntax

Syntax:〈Bindings〉 should have the form

((〈variable1〉 〈init1〉) . . . ),

where each〈init〉 is an expression, and〈body〉 should be a sequence of one or more expres-
sions. It is an error for a〈variable〉 to appear more than once in the list of variables being
bound.

Semantics:The〈init〉s are evaluated in the current environment (in some unspecified order),
the 〈variable〉s are bound to fresh locations holding the results, the〈body〉 is evaluated in
the extended environment, and the value(s) of the last expression of〈body〉 is(are) returned.
Each binding of a〈variable〉 has〈body〉 as its region.

(let ((x 2) (y 3))
(* x y)) H⇒ 6

(let ((x 2) (y 3))
(let ((x 7)

(z(+ x y)))
(* z x))) H⇒ 35

See also namedlet, Section 5.2.4.

(let* 〈bindings〉 〈body〉) library syntax

Syntax:〈Bindings〉 should have the form

((〈variable1〉 〈init1〉) . . . ),

and〈body〉 should be a sequence of one or more expressions.

Semantics:Let* is similar tolet, but the bindings are performed sequentially from left
to right, and the region of a binding indicated by(〈variable〉 〈init〉) is that part of thelet*
expression to the right of the binding. Thus the second binding is done in an environment
in which the first binding is visible, and so on.

(let ((x 2) (y 3))
(let* ((x 7)

(z (+ x y)))
(* z x))) H⇒ 70

(letrec 〈bindings〉 〈body〉) library syntax

Syntax:〈Bindings〉 should have the form

((〈variable1〉 〈init1〉) . . . ),

and〈body〉 should be a sequence of one or more expressions. It is an error for a〈variable〉
to appear more than once in the list of variables being bound.

Semantics: The 〈variable〉s are bound to fresh locations holding undefined values, the
〈init〉s are evaluated in the resulting environment (in some unspecified order), each〈variable〉

is assigned to the result of the corresponding〈init〉, the〈body〉 is evaluated in the resulting
environment, and the value(s) of the last expression in〈body〉 is(are) returned. Each binding
of a 〈variable〉 has the entireletrec expression as its region, making it possible to define
mutually recursive procedures.

(letrec ((even?
(lambda (n)
(if (zero? n)

#t
(odd? (- n 1)))))

(odd?
(lambda (n)
(if (zero? n)

#f
(even? (- n 1))))))

(even? 88)) H⇒ #t

One restriction onletrec is very important: it must be possible to evaluate each〈init〉
without assigning or referring to the value of any〈variable〉. If this restriction is violated,
then it is an error. The restriction is necessary because Scheme passes arguments by value
rather than by name. In the most common uses ofletrec, all the 〈init〉s arelambda
expressions and the restriction is satisfied automatically.

5.2.3. Sequencing.

(begin 〈expression1〉 〈expression2〉 . . .) library syntax

The〈expression〉s are evaluated sequentially from left to right, and the value(s) of the last
〈expression〉 is(are) returned. This expression type is used to sequence side effects such as
input and output.

(define x 0)

(begin (set! x 5)
(+ x 1)) H⇒ 6

(begin (display "4 plus 1 equals ")
(display (+ 4 1))) H⇒ unspecified

and prints4 plus 1 equals 5

5.2.4. Iteration.

(do ((〈variable1〉 〈init1〉 〈step1〉) library syntax
. . .)
(〈test〉 〈expression〉 . . .)
〈command〉 . . .)

Do is an iteration construct. It specifies a set of variables to be bound, how they are to be
initialized at the start, and how they are to be updated on each iteration. When a termination
condition is met, the loop exits after evaluating the〈expression〉s.
Do expressions are evaluated as follows: The〈init〉 expressions are evaluated (in some

unspecified order), the〈variable〉s are bound to fresh locations, the results of the〈init〉
expressions are stored in the bindings of the〈variable〉s, and then the iteration phase begins.

Each iteration begins by evaluating〈test〉; if the result is false (see Section 7.3.1), then
the 〈command〉 expressions are evaluated in order for effect, the〈step〉 expressions are
evaluated in some unspecified order, the〈variable〉s are bound to fresh locations, the results
of the〈step〉s are stored in the bindings of the〈variable〉s, and the next iteration begins.

If 〈test〉 evaluates to a true value, then the〈expression〉s are evaluated from left to right
and the value(s) of the last〈expression〉 is(are) returned. If no〈expression〉s are present,
then the value of thedo expression is unspecified.

The region of the binding of a〈variable〉 consists of the entiredo expression except for
the〈init〉s. It is an error for a〈variable〉 to appear more than once in the list ofdo variables.

A 〈step〉 may be omitted, in which case the effect is the same as if(〈variable〉〈init〉
〈variable〉) had been written instead of(〈variable〉 〈init〉).

(do ((vec (make-vector 5))
(i 0 (+ i 1)))
((= i 5) vec)

(vector-set! vec i i)) H⇒ #(0 1 2 3 4)

(let ((x ’(1 3 5 7 9)))
(do ((x x (cdr x))

(sum 0 (+ sum (car x))))
((null? x) sum))) H⇒ 25

(let 〈variable〉 〈bindings〉 〈body〉) library syntax

“Namedlet” is a variant on the syntax oflet which provides a more general looping
construct thando and may also be used to express recursions. It has the same syntax and
semantics as ordinarylet except that〈variable〉 is bound within〈body〉 to a procedure
whose formal arguments are the bound variables and whose body is〈body〉. Thus the
execution of〈body〉 may be repeated by invoking the procedure named by〈variable〉.

(let loop ((numbers ’(3 -2 1 6 -5)) (nonneg ’())(neg ’()))
(cond ((null? numbers) (list nonneg neg))

((>= (car numbers) 0)
(loop (cdr numbers)

(cons (car numbers) nonneg)
neg))

((< (car numbers) 0)
(loop (cdr numbers)

nonneg
(cons (car numbers) neg)))))

H⇒ ((6 1 3) (-5 -2))

5.2.5. Delayed evaluation.

(delay 〈expression〉) library syntax

Thedelay construct is used together with the procedureforce to implementlazy eval-
uationor call by need. (delay 〈expression〉) returns an object called apromisewhich at
some point in the future may be asked (by theforce procedure) to evaluate〈expression〉,
and deliver the resulting value. The effect of〈expression〉 returning multiple values is
unspecified.

See the description offorce (Section 7.4) for a more complete description ofdelay.

5.2.6. Quasiquotation.

(quasiquote 〈qq template〉) syntax
`〈qq template〉 syntax

“Backquote” or “quasiquote” expressions are useful for constructing a list or vector
structure when most but not all of the desired structure is known in advance. If no commas
appear within the〈qq template〉, the result of evaluating’〈qq template〉 is equivalent to the
result of evaluating’〈qq template〉. If a comma appears within the〈qq template〉, however,
the expression following the comma is evaluated (“unquoted”) and its result is inserted
into the structure instead of the comma and the expression. If a comma appears followed
immediately by an at-sign, (@), then the following expression must evaluate to a list; the
opening and closing parentheses of the list are then “stripped away” and the elements of
the list are inserted in place of the comma at-sign expression sequence. A comma at-sign
should only appear within a list or vector〈qq template〉.

`(list ,(+ 1 2) 4) H⇒ (list 3 4)
(let ((name ’a)) `(list ,name ’,name))

H⇒ (list a (quote a))
`(a ,(+ 1 2) ,@(map abs ’(4 -5 6)) b)

H⇒ (a 3 4 5 6 b)
`((foo ,(- 10 3)),@(cdr ’(c)) . ,(car ’(cons)))

H⇒ ((foo 7) . cons)
`#(10 5 ,(sqrt 4) ,@(map sqrt ’(16 9)) 8)

H⇒ #(10 5 2 4 3 8)

Quasiquote forms may be nested. Substitutions are made only for unquoted compo-
nents appearing at the same nesting level as the outermost backquote. The nesting level
increases by one inside each successive quasiquotation, and decreases by one inside each
unquotation.

`(a `(b ,(+ 1 2) ,(foo ,(+ 1 3)d) e) f)
H⇒ (a `(b ,(+ 1 2)
,(foo 4 d) e) f)

(let ((name1 ’x)
(name2 ’y))

`(a `(b ,,name1 ,’,name2 d) e)) H⇒ (a `(b ,x ,’y d) e)

The two notations̀ 〈qq template〉 and(quasiquote 〈qq template〉) are identical in
all respects.,〈expression〉 is identical to(unquote 〈expression〉), and ,@〈expression〉
is identical to(unquote-splicing 〈expression〉). The external syntax generated by
write for two-element lists whose car is one of these symbols may vary between imple-
mentations.

(quasiquote (list (unquote (+ 1 2)) 4))
H⇒ (list 3 4)

’(quasiquote (list (unquote (+ 1 2)) 4))
H⇒ `(list ,(+ 1 2) 4)

i.e.,(quasiquote (list (unquote (+ 1 2)) 4))

Unpredictable behavior can result if any of the symbols quasiquote, unquote, or
unquote-splicing appear in positions within a〈qq template〉 otherwise than as described
above.

5.3. Macros

Scheme programs can define and use new derived expression types, calledmacros. Program-
defined expression types have the syntax

(〈keyword〉 〈datum〉 . . .)

where〈keyword〉 is an identifier that uniquely determines the expression type. This identifier
is called thesyntactic keyword, or simply keyword, of the macro. The number of the
〈datum〉s, and their syntax, depends on the expression type.

Each instance of a macro is called auseof the macro. The set of rules that specifies how
a use of a macro is transcribed into a more primitive expression is called thetransformer
of the macro.

The macro definition facility consists of two parts:

• A set of expressions used to establish that certain identifiers are macro keywords, associate
them with macro transformers, and control the scope within which a macro is defined,
and
• a pattern language for specifying macro transformers.

The syntactic keyword of a macro may shadow variable bindings, and local variable
bindings may shadow keyword bindings. All macros defined using the pattern language
are “hygienic” and “referentially transparent” and thus preserve Scheme’s lexical scoping
[2, 7, 9, 14, 15]:

• If a macro transformer inserts a binding for an identifier (variable or keyword), the
identifier will in effect be renamed throughout its scope to avoid conflicts with other
identifiers. Note that adefine at top level may or may not introduce a binding; see
Section 6.2.
• If a macro transformer inserts a free reference to an identifier, the reference refers to

the binding that was visible where the transformer was specified, regardless of any local
bindings that may surround the use of the macro.

5.3.1. Binding constructs for syntactic keywords.Let-syntax andletrec-syntax
are analogous tolet andletrec, but they bind syntactic keywords to macro transformers
instead of binding variables to locations that contain values. Syntactic keywords may also
be bound at top level; see Section 6.3.

(let-syntax 〈bindings〉 〈body〉) syntax

Syntax:〈Bindings〉 should have the form

((〈keyword〉 〈transformer spec〉) . . .)
Each〈keyword〉 is an identifier, each〈transformer spec〉 is an instance ofsyntax-rules,
and〈body〉 should be a sequence of one or more expressions. It is an error for a〈keyword〉
to appear more than once in the list of keywords being bound.

Semantics:The 〈body〉 is expanded in the syntactic environment obtained by extending
the syntactic environment of thelet-syntax expression with macros whose keywords
are the〈keyword〉s, bound to the specified transformers. Each binding of a〈keyword〉 has
〈body〉 as its region.

(let-syntax ((when (syntax-rules ()
((when test stmt1 stmt2 ...)
(if test

(begin stmt1
stmt2 ...))))))

(let ((if #t))
(when if (set! if ’now))
if)) H⇒ now

(let ((x ’outer))
(let-syntax ((m (syntax-rules () ((m)x))))
(let ((x ’inner))
(m)))) H⇒ outer

(letrec-syntax 〈bindings〉 〈body〉) syntax

Syntax:Same as forlet-syntax.

Semantics:The 〈body〉 is expanded in the syntactic environment obtained by extending
the syntactic environment of theletrec-syntax expression with macros whose keywords
are the〈keyword〉s, bound to the specified transformers. Each binding of a〈keyword〉 has
the 〈bindings〉 as well as the〈body〉 within its region, so the transformers can transcribe
expressions into uses of the macros introduced by theletrec-syntax expression.

(letrec-syntax
((my-or (syntax-rules ()

((my-or) #f)
((my-or e) e)
((my-or e1 e2 ...)
(let ((temp e1))
(if temp

temp
(my-or e2 ...)))))))

(let ((x #f)
(y 7)
(temp 8)
(let odd?)
(if even?))

(my-or x
(let temp)
(if y)
y))) H⇒ 7

5.3.2. Pattern language. A 〈transformer spec〉 has the following form:

(syntax-rules 〈literals〉 〈syntax rule〉 . . .)

Syntax:〈Literals〉 is a list of identifiers and each〈syntax rule〉 should be of the form

(〈pattern〉 〈template〉)

The〈pattern〉 in a〈syntax rule〉 is a list〈pattern〉 that begins with the keyword for the macro.
A 〈pattern〉 is either an identifier, a constant, or one of the following

(〈pattern〉 . . . )
(〈pattern〉 〈pattern〉 . . . . 〈pattern〉)
(〈pattern〉 . . . 〈pattern〉 〈ellipsis〉)
#(〈pattern〉 . . .)
#(〈pattern〉 . . . 〈pattern〉 〈ellipsis〉)

and a template is either an identifier, a constant, or one of the following

(〈element〉 . . .)
(〈element〉 〈element〉 . . . . 〈template〉)
#(〈element〉 . . .)

where an〈element〉 is a〈template〉 optionally followed by an〈ellipsis〉 and an〈ellipsis〉 is
the identifier “...” (which cannot be used as an identifier in either a template or a pattern).

Semantics:An instance ofsyntax-rulesproduces a new macro transformer by specifying
a sequence of hygienic rewrite rules. A use of a macro whose keyword is associated with
a transformer specified bysyntax-rules is matched against the patterns contained in the
〈syntax rule〉s, beginning with the leftmost〈syntax rule〉. When a match is found, the macro
use is transcribed hygienically according to the template.

An identifier that appears in the pattern of a〈syntax rule〉 is apattern variable, unless it is
the keyword that begins the pattern, is listed in〈literals〉, or is the identifier “...”. Pattern
variables match arbitrary input elements and are used to refer to elements of the input in the
template. It is an error for the same pattern variable to appear more than once in a〈pattern〉.

The keyword at the beginning of the pattern in a〈syntax rule〉 is not involved in the
matching and is not considered a pattern variable or literal identifier.

Rationale:The scope of the keyword is determined by the expression or syntax definition
that binds it to the associated macro transformer. If the keyword were a pattern variable
or literal identifier, then the template that follows the pattern would be within its scope
regardless of whether the keyword were bound bylet-syntax or byletrec-syntax.

Identifiers that appear in〈literals〉are interpreted as literal identifiers to be matched against
corresponding subforms of the input. A subform in the input matches a literal identifier if
and only if it is an identifier and either both its occurrence in the macro expression and its
occurrence in the macro definition have the same lexical binding, or the two identifiers are
equal and both have no lexical binding.

A subpattern followed by. . . can match zero or more elements of the input. It is an error
for . . . to appear in〈literals〉. Within a pattern the identifier. . . must follow the last element
of a nonempty sequence of subpatterns.

More formally, an input formF matches a patternP if and only if:

• P is a non-literal identifier; or
• P is a literal identifier andF is an identifier with the same binding; or
• P is a list(P1 . . . Pn) andF is a list ofn forms that matchP1 throughPn, respectively;

or
• P is an improper list(P1 P2 . . . Pn . Pn+1) andF is a list or improper list ofn or more

forms that matchP1 throughPn, respectively, and whosenth “cdr” matchesPn+1; or
• P is of the form (P1 . . . Pn Pn+1 〈ellipsis〉) where〈ellipsis〉 is the identifier... andF

is a proper list of at leastn forms, the firstn of which matchP1 throughPn, respectively,
and each remaining element ofF matchesPn+1; or

• P is a vector of the form#(P1 . . . Pn) andF is a vector ofn forms that matchP1 through
Pn; or
• P is of the form#(P1 . . . Pn Pn+1 〈ellipsis〉) where〈ellipsis〉 is the identifier... andF

is a vector ofn or more forms the firstn of which matchP1 throughPn, respectively, and
each remaining element ofF matchesPn+1; or
• P is a datum andF is equal toP in the sense of theequal? procedure.

It is an error to use a macro keyword, within the scope of its binding, in an expression
that does not match any of the patterns.

When a macro use is transcribed according to the template of the matching〈syntax rule〉,
pattern variables that occur in the template are replaced by the subforms they match in the
input. Pattern variables that occur in subpatterns followed by one or more instances of
the identifier. . . are allowed only in subtemplates that are followed by as many instances
of . . . . They are replaced in the output by all of the subforms they match in the input,
distributed as indicated. It is an error if the output cannot be built up as specified.

Identifiers that appear in the template but are not pattern variables or the identifier...
are inserted into the output as literal identifiers. If a literal identifier is inserted as a free
identifier then it refers to the binding of that identifier within whose scope the instance of
syntax-rules appears. If a literal identifier is inserted as a bound identifier then it is in
effect renamed to prevent inadvertent captures of free identifiers.

As an example, iflet andcond are defined as in Section 8.3 then they are hygienic (as
required) and the following is not an error.

(let ((=> #f))
(cond (#t => ’ok))) H⇒ ok

The macro transformer forcond recognizes=> as a local variable, and hence an expres-
sion, and not as the top-level identifier=>, which the macro transformer treats as a syntactic
keyword. Thus the example expands into

(let ((=> #f))
(if #t (begin => ’ok)))

instead of

(let ((=> #f))
(let ((temp #t))
(if temp (’ok temp))))

which would result in an invalid procedure call.

6. Program structure

6.1. Programs

A Scheme program consists of a sequence of expressions, definitions, and syntax definitions.
Expressions are described in Section 5; definitions and syntax definitions are the subject of
the rest of the present section.

Programs are typically stored in files or entered interactively to a running Scheme system,
although other paradigms are possible; questions of user interface lie outside the scope of
this report. (Indeed, Scheme would still be useful as a notation for expressing computational
methods even in the absence of a mechanical implementation.)

Definitions and syntax definitions occurring at the top level of a program can be interpreted
declaratively. They cause bindings to be created in the top level environment or modify the
value of existing top-level bindings. Expressions occurring at the top level of a program are
interpreted imperatively; they are executed in order when the program is invoked or loaded,
and typically perform some kind of initialization.

At the top level of a program(begin 〈form1〉 . . .) is equivalent to the sequence of
expressions, definitions, and syntax definitions that form the body of thebegin.

6.2. Definitions

Definitions are valid in some, but not all, contexts where expressions are allowed. They are
valid only at the top level of a〈program〉 and at the beginning of a〈body〉.

A definition should have one of the following forms:

• (define 〈variable〉 〈expression〉)
• (define (〈variable〉 〈formals〉) 〈body〉)

〈Formals〉 should be either a sequence of zero or more variables, or a sequence of one or
more variables followed by a space-delimited period and another variable (as in a lambda
expression). This form is equivalent to

(define 〈variable〉
(lambda (〈formals〉) 〈body〉)).

• (define (〈variable〉 . 〈formals〉) 〈body〉)
〈Formals〉 should be a single variable. This form is equivalent to

(define 〈variable〉
(lambda 〈formal〉 〈body〉)).

6.2.1. Top level definitions. At the top level of a program, a definition

(define 〈variable〉 〈expression〉)

has essentially the same effect as the assignment expression

(set! 〈variable〉 〈expression〉)

if 〈variable〉 is bound. If〈variable〉 is not bound, however, then the definition will bind
〈variable〉 to a new location before performing the assignment, whereas it would be an error
to perform aset! on an unbound variable.

(define add3
(lambda (x) (+ x 3)))

(add3 3) H⇒ 6
(define first car)
(first ’(1 2)) H⇒ 1

Some implementations of Scheme use an initial environment in which all possible vari-
ables are bound to locations, most of which contain undefined values. Top level definitions
in such an implementation are truly equivalent to assignments.

6.2.2. Internal definitions. Definitions may occur at the beginning of a〈body〉 (that is,
the body of alambda, let, let*, letrec, let-syntax, or letrec-syntax expression
or that of a definition of an appropriate form). Such definitions are known asinternal
definitionsas opposed to the top level definitions described above. The variable defined by
an internal definition is local to the〈body〉. That is,〈variable〉 is bound rather than assigned,
and the region of the binding is the entire〈body〉. For example,

(let ((x 5))
(define foo (lambda (y) (bar x y)))
(define bar (lambda (a b) (+ (* a b) a)))
(foo (+ x 3))) H⇒ 45

A 〈body〉containing internal definitions can always be converted into a completely equiva-
lentletrecexpression. For example, theletexpression in the above example is equivalent
to

(let ((x 5))
(letrec ((foo (lambda (y) (bar x y)))

(bar (lambda (a b) (+ (* a b) a))))
(foo (+ x 3))))

Just as for the equivalentletrec expression, it must be possible to evaluate each
〈expression〉 of every internal definition in a〈body〉 without assigning or referring to the
value of any〈variable〉 being defined.

Wherever an internal definition may occur(begin 〈definition1〉 . . .) is equivalent to the
sequence of definitions that form the body of thebegin.

6.3. Syntax definitions

Syntax definitions are valid only at the top level of a〈program〉 . They have the following
form:

(define-syntax 〈keyword〉 〈transformer spec〉)

〈keyword〉 is an identifier, and the〈transformer spec〉 should be an instance ofsyntax-
rules. The top-level syntactic environment is extended by binding the〈keyword〉 to the
specified transformer.

There is nodefine-syntax analogue of internal definitions.
Although macros may expand into definitions and syntax definitions in any context that

permits them, it is an error for a definition or syntax definition to shadow a syntactic
keyword whose meaning is needed to determine whether some form in the group of forms
that contains the shadowing definition is in fact a definition, or, for internal definitions, is
needed to determine the boundary between the group and the expressions that follow the
group. For example, the following are errors:

(define define 3)

(begin (define begin list))

(let-syntax
((foo (syntax-rules ()

((foo (proc args ...) body ...)
(define proc
(lambda (args ...)
body ...))))))

(let ((x 3))
(foo (plus x y) (+ x y))
(define foo x)
(plus foo x)))

7. Standard procedures

This section describes Scheme’s built-in procedures. The initial (or “top level”) Scheme
environment starts out with a number of variables bound to locations containing useful
values, most of which are primitive procedures that manipulate data. For example, the
variableabs is bound to (a location initially containing) a procedure of one argument that
computes the absolute value of a number, and the variable+ is bound to a procedure that
computes sums. Built-in procedures that can easily be written in terms of other built-in
procedures are identified as “library procedures”.

A program may use a top-level definition to bind any variable. It may subsequently alter
any such binding by an assignment (see Section 5.1.6). These operations do not modify
the behavior of Scheme’s built-in procedures. Altering any top-level binding that has not
been introduced by a definition has an unspecified effect on the behavior of the built-in
procedures.

7.1. Equivalence predicates

A predicateis a procedure that always returns a boolean value (#t or #f). An equivalence
predicateis the computational analogue of a mathematical equivalence relation (it is sym-
metric, reflexive, and transitive). Of the equivalence predicates described in this section,

eq? is the finest or most discriminating, andequal? is the coarsest.Eqv? is slightly less
discriminating thaneq?.

(eqv? obj1 obj2) procedure

Theeqv? procedure defines a useful equivalence relation on objects. Briefly, it returns
# t if obj1 andobj2 should normally be regarded as the same object. This relation is left
slightly open to interpretation, but the following partial specification ofeqv? holds for all
implementations of Scheme.

Theeqv? procedure returns#t if:

• obj1 andobj2 are both#t or both#f.
• obj1 andobj2 are both symbols and

(string=? (symbol->string obj1)
(symbol->string obj2)) H⇒ #t

Note: This assumes that neitherobj1 nor obj2 is an “uninterned symbol” as alluded
to in Section 7.3.3. This report does not presume to specify the behavior ofeqv? on
implementation-dependent extensions.

• obj1 andobj2 are both numbers, are numerically equal (see=, Section 7.2), and are either
both exact or both inexact.
• obj1 andobj2 are both characters and are the same character according to thechar=?

procedure (Section 7.3.4).
• bothobj1 andobj2 are the empty list.
• obj1 andobj2 are pairs, vectors, or strings that denote the same locations in the store

(Section 4.4).
• obj1 andobj2 are procedures whose location tags are equal (Section 5.1.4).

Theeqv? procedure returns#f if:

• obj1 andobj2 are of different types (Section 4.2).
• one ofobj1 andobj2 is #t but the other is#f.
• obj1 andobj2 are symbols but

(string=? (symbol->string obj1)
(symbol->string obj2)) H⇒ #f

• one ofobj1 andobj2 is an exact number but the other is an inexact number.
• obj1 andobj2 are numbers for which the= procedure returns#f.
• obj1 andobj2 are characters for which thechar=? procedure returns#f.
• one ofobj1 andobj2 is the empty list but the other is not.
• obj1 andobj2 are pairs, vectors, or strings that denote distinct locations.

• obj1 andobj2 are procedures that would behave differently (return different value(s) or
have different side effects) for some arguments.

(eqv? ’a ’a) H⇒ #t
(eqv? ’a ’b) H⇒ #f
(eqv? 2 2) H⇒ #t
(eqv? ’() ’()) H⇒ #t
(eqv? 100000000 100000000) H⇒ #t
(eqv? (cons 1 2) (cons 1 2)) H⇒ #f
(eqv? (lambda () 1)

(lambda () 2)) H⇒ #f
(eqv? #f ’nil) H⇒ #f
(let ((p (lambda (x) x)))
(eqv? p p)) H⇒ #t

The following examples illustrate cases in which the above rules do not fully specify the
behavior ofeqv?. All that can be said about such cases is that the value returned byeqv?
must be a boolean.

(eqv? "" "") H⇒ unspecified
(eqv? ’#() ’#()) H⇒ unspecified
(eqv? (lambda (x) x)

(lambda (x) x)) H⇒ unspecified
(eqv? (lambda (x) x)

(lambda (y) y)) H⇒ unspecified

The next set of examples shows the use ofeqv? with procedures that have local state.
Gen-countermust return a distinct procedure every time, since each procedure has its own
internal counter.Gen-loser, however, returns equivalent procedures each time, since the
local state does not affect the value or side effects of the procedures.

(define gen-counter
(lambda ()
(let ((n 0))
(lambda () (set! n (+ n 1)) n))))

(let ((g (gen-counter)))
(eqv? g g)) H⇒ #t

(eqv? (gen-counter) (gen-counter)) H⇒ #f
(define gen-loser
(lambda ()
(let ((n 0))
(lambda () (set! n (+ n 1)) 27))))

(let ((g (gen-loser)))
(eqv? g g)) H⇒ #t

(eqv? (gen-loser) (gen-loser)) H⇒ unspecified

(letrec ((f (lambda () (if (eqv? f g) ’both ’f)))
(g (lambda () (if (eqv? f g) ’both ’g))))

(eqv? f g)) H⇒ unspecified
(letrec ((f (lambda () (if (eqv? f g) ’f ’both)))

(g (lambda () (if (eqv? f g) ’g ’both))))
(eqv? f g)) H⇒ #f

Since it is an error to modify constant objects (those returned by literal expressions),
implementations are permitted, though not required, to share structure between constants
where appropriate. Thus the value ofeqv? on constants is sometimes implementation-
dependent.

(eqv? ’(a) ’(a)) H⇒ unspecified
(eqv? "a" "a") H⇒ unspecified
(eqv? ’(b) (cdr ’(a b))) H⇒ unspecified
(let ((x ’(a)))
(eqv? x x)) H⇒ #t

Rationale:The above definition ofeqv? allows implementations latitude in their treatment
of procedures and literals: implementations are free either to detect or to fail to detect that
two procedures or two literals are equivalent to each other, and can decide whether or not
to merge representations of equivalent objects by using the same pointer or bit pattern to
represent both.

(eq? obj1 obj2) procedure

Eq? is similar toeqv? except that in some cases it is capable of discerning distinctions
finer than those detectable byeqv?.
Eq? andeqv? are guaranteed to have the same behavior on symbols, booleans, the empty

list, pairs, procedures, and nonempty strings and vectors.Eq?’s behavior on numbers and
characters is implementation-dependent, but it will always return either true or false, and
will return true only wheneqv? would also return true.Eq? may also behave differently
from eqv? on empty vectors and empty strings.

(eq? ’a ’a) H⇒ #t
(eq? ’(a) ’(a)) H⇒ unspecified
(eq? (list ’a) (list ’a)) H⇒ #f
(eq? "a" "a") H⇒ unspecified
(eq? "" "") H⇒ unspecified
(eq? ’() ’()) H⇒ #t
(eq? 2 2) H⇒ unspecified
(eq? #\A #\A) H⇒ unspecified
(eq? car car) H⇒ #t
(let ((n (+2 3)))
(eq? n n)) H⇒ unspecified

(let ((x ’(a)))

(eq? x x)) H⇒ #t
(let ((x ’#()))
(eq? x x)) H⇒ #t

(let ((p (lambda (x) x)))
(eq? p p)) H⇒ #t

Rationale: It will usually be possible to implementeq? much more efficiently thaneqv?,
for example, as a simple pointer comparison instead of as some more complicated operation.
One reason is that it may not be possible to computeeqv? of two numbers in constant time,
whereaseq? implemented as pointer comparison will always finish in constant time.Eq?
may be used likeeqv? in applications using procedures to implement objects with state
since it obeys the same constraints aseqv?.

(equal? obj1 obj2) library procedure

Equal? recursively compares the contents of pairs, vectors, and strings, applyingeqv?
on other objects such as numbers and symbols. A rule of thumb is that objects are generally
equal? if they print the same.Equal? may fail to terminate if its arguments are circular
data structures.

(equal? ’a ’a) H⇒ #t
(equal? ’(a)’(a)) H⇒ #t
(equal? ’(a (b) c)

’(a (b) c)) H⇒ #t
(equal? "abc" "abc") H⇒ #t
(equal? 2 2) H⇒ #t
(equal? (make-vector 5 ’a)

(make-vector 5 ’a)) H⇒ #t
(equal? (lambda (x) x)

(lambda (y) y)) H⇒ unspecified

7.2. Numbers

Numerical computation has traditionally been neglected by the Lisp community. Until Com-
mon Lisp there was no carefully thought out strategy for organizing numerical
computation, and with the exception of the MacLisp system [20] little effort was made
to execute numerical code efficiently. This report recognizes the excellent work of the
Common Lisp committee and accepts many of their recommendations. In some ways this
report simplifies and generalizes their proposals in a manner consistent with the purposes
of Scheme.

It is important to distinguish between the mathematical numbers, the Scheme numbers
that attempt to model them, the machine representations used to implement the Scheme
numbers, and notations used to write numbers. This report uses the typesnumber, complex,
real, rational, and integer to refer to both mathematical numbers and Scheme numbers.

Machine representations such as fixed point and floating point are referred to by names
such asfixnumandflonum.

7.2.1. Numerical types. Mathematically, numbers may be arranged into a tower of sub-
types in which each level is a subset of the level above it:

number
complex
real
rational
integer

For example, 3 is an integer. Therefore, 3 is also a rational, a real, and a complex. The
same is true of the Scheme numbers that model 3. For Scheme numbers, these types are
defined by the predicatesnumber?, complex?, real?, rational?, andinteger?.

There is no simple relationship between a number’s type and its representation inside
a computer. Although most implementations of Scheme will offer at least two different
representations of 3, these different representations denote the same integer.

Scheme’s numerical operations treat numbers as abstract data, as independent of their
representation as possible. Although an implementation of Scheme may use fixnum, flonum,
and perhaps other representations for numbers, this should not be apparent to a casual
programmer writing simple programs.

It is necessary, however, to distinguish between numbers that are represented exactly
and those that may not be. For example, indexes into data structures must be known
exactly, as must some polynomial coefficients in a symbolic algebra system. On the other
hand, the results of measurements are inherently inexact, and irrational numbers may be
approximated by rational and therefore inexact approximations. In order to catch uses of
inexact numbers where exact numbers are required, Scheme explicitly distinguishes exact
from inexact numbers. This distinction is orthogonal to the dimension of type.

7.2.2. Exactness. Scheme numbers are eitherexactor inexact. A number is exact if it was
written as an exact constant or was derived from exact numbers using only exact operations.
A number is inexact if it was written as an inexact constant, if it was derived using inexact
ingredients, or if it was derived using inexact operations. Thus inexactness is a contagious
property of a number.

If two implementations produce exact results for a computation that did not involve
inexact intermediate results, the two ultimate results will be mathematically equivalent. This
is generally not true of computations involving inexact numbers since approximate methods
such as floating point arithmetic may be used, but it is the duty of each implementation to
make the result as close as practical to the mathematically ideal result.

Rational operations such as+ should always produce exact results when given exact
arguments. If the operation is unable to produce an exact result, then it may either report
the violation of an implementation restriction or it may silently coerce its result to an inexact
value. See Section 7.2.3.



P1: VBI

HIGHER-ORDER AND SYMBOLIC COMPUTATION KL610-01-Abelson August 6, 1998 17:15

42 ABELSON ET AL.

With the exception ofinexact->exact, the operations described in this section must
generally return inexact results when given any inexact arguments. An operation may,
however, return an exact result if it can prove that the value of the result is unaffected by
the inexactness of its arguments. For example, multiplication of any number by an exact
zero may produce an exact zero result, even if the other argument is inexact.

7.2.3. Implementation restrictions. Implementations of Scheme are not required to im-
plement the whole tower of subtypes given in Section 7.2.1, but they must implement a
coherent subset consistent with both the purposes of the implementation and the spirit of
the Scheme language. For example, an implementation in which all numbers are real may
still be quite useful.

Implementations may also support only a limited range of numbers of any type, subject
to the requirements of this section. The supported range for exact numbers of any type
may be different from the supported range for inexact numbers of that type. For example,
an implementation that uses flonums to represent all its inexact real numbers may support
a practically unbounded range of exact integers and rationals while limiting the range of
inexact reals (and therefore the range of inexact integers and rationals) to the dynamic range
of the flonum format. Furthermore the gaps between the representable inexact integers and
rationals are likely to be very large in such an implementation as the limits of this range are
approached.

An implementation of Scheme must support exact integers throughout the range of num-
bers that may be used for indexes of lists, vectors, and strings or that may result from
computing the length of a list, vector, or string. Thelength, vector-length, and
string-length procedures must return an exact integer, and it is an error to use any-
thing but an exact integer as an index. Furthermore, any integer constant within the index
range, if expressed by an exact integer syntax, will indeed be read as an exact integer,
regardless of any implementation restrictions that may apply outside this range. Finally,
the procedures listed below will always return an exact integer result provided all their
arguments are exact integers and the mathematically expected result is representable as an
exact integer within the implementation:

+ - *
quotient remainder modulo
max min abs
numerator denominator gcd
lcm floor ceiling
truncate round rationalize
expt

Implementations are encouraged, but not required, to support exact integers and exact
rationals of practically unlimited size and precision, and to implement the above proce-
dures and the/ procedure in such a way that they always return exact results when given
exact arguments. If one of these procedures is unable to deliver an exact result when given
exact arguments, then it may either report a violation of an implementation restriction or
it may silently coerce its result to an inexact number. Such a coercion may cause an error
later.

An implementation may use floating point and other approximate representation strate-
gies for inexact numbers. This report recommends, but does not require, that the IEEE 32-bit
and 64-bit floating point standards be followed by implementations that use flonum repre-
sentations, and that implementations using other representations should match or exceed
the precision achievable using these floating point standards [12].

In particular, implementations that use flonum representations must follow these rules: A
flonum result must be represented with at least as much precision as is used to express any
of the inexact arguments to that operation. It is desirable (but not required) for potentially
inexact operations such assqrt, when applied to exact arguments, to produce exact answers
whenever possible (for example the square root of an exact 4 ought to be an exact 2). If,
however, an exact number is operated upon so as to produce an inexact result (as bysqrt),
and if the result is represented as a flonum, then the most precise flonum format available
must be used; but if the result is represented in some other way then the representation must
have at least as much precision as the most precise flonum format available.

Although Scheme allows a variety of written notations for numbers, any particular im-
plementation may support only some of them. For example, an implementation in which all
numbers are real need not support the rectangular and polar notations for complex numbers.
If an implementation encounters an exact numerical constant that it cannot represent as an
exact number, then it may either report a violation of an implementation restriction or it
may silently represent the constant by an inexact number.

7.2.4. Syntax of numerical constants.The syntax of the written representations for num-
bers is described formally in Section 8.1.1. Note that case is not significant in numerical
constants.

A number may be written in binary, octal, decimal, or hexadecimal by the use of a radix
prefix. The radix prefixes are#b (binary),#o (octal),#d (decimal), and#x (hexadecimal).
With no radix prefix, a number is assumed to be expressed in decimal.

A numerical constant may be specified to be either exact or inexact by a prefix. The
prefixes are#e for exact, and#i for inexact. An exactness prefix may appear before or after
any radix prefix that is used. If the written representation of a number has no exactness
prefix, the constant may be either inexact or exact. It is inexact if it contains a decimal
point, an exponent, or a “#” character in the place of a digit, otherwise it is exact.

In systems with inexact numbers of varying precisions it may be useful to specify the
precision of a constant. For this purpose, numerical constants may be written with an
exponent marker that indicates the desired precision of the inexact representation. The letters
s,f,d, andlspecify the use ofshort, single, double, andlongprecision, respectively. (When
fewer than four internal inexact representations exist, the four size specifications are mapped
onto those available. For example, an implementation with two internal representations
may map short and single together and long and double together.) In addition, the exponent
markere specifies the default precision for the implementation. The default precision has
at least as much precision asdouble, but implementations may wish to allow this default to
be set by the user.

3.14159265358979F0
Round to single —3.141593

0.6L0
Extend to long —.600000000000000

7.2.5. Numerical operations. The reader is referred to Section 2.3.3 for a summary of
the naming conventions used to specify restrictions on the types of arguments to numerical
routines. The examples used in this section assume that any numerical constant written
using an exact notation is indeed represented as an exact number. Some examples also
assume that certain numerical constants written using an inexact notation can be represented
without loss of accuracy; the inexact constants were chosen so that this is likely to be true
in implementations that use flonums to represent inexact numbers.

(number? obj) procedure
(complex? obj) procedure
(real? obj) procedure
(rational? obj) procedure
(integer? obj) procedure

These numerical type predicates can be applied to any kind of argument, including non-
numbers. They return#t if the object is of the named type, and otherwise they return#f.
In general, if a type predicate is true of a number then all higher type predicates are also
true of that number. Consequently, if a type predicate is false of a number, then all lower
type predicates are also false of that number.

If z is an inexact complex number, then(real? z) is true if and only if(zero? (imag-
part z)) is true. Ifx is an inexact real number, then(integer? x) is true if and only if
(= x (round x)).

(complex? 3+4i) H⇒ #t
(complex? 3) H⇒ #t
(real? 3) H⇒ #t
(real? -2.5+0.0i) H⇒ #t
(real? #e1e10) H⇒ #t
(rational? 6/10) H⇒ #t
(rational? 6/3) H⇒ #t
(integer? 3+0i) H⇒ #t
(integer? 3.0) H⇒ #t
(integer? 8/4) H⇒ #t

Note: The behavior of these type predicates on inexact numbers is unreliable, since any
inaccuracy may affect the result.

Note: In many implementations therational? procedure will be the same asreal?,
and thecomplex? procedure will be the same asnumber?, but unusual implementations
may be able to represent some irrational numbers exactly or may extend the number system
to support some kind of noncomplex numbers.



P1: VBI

HIGHER-ORDER AND SYMBOLIC COMPUTATION KL610-01-Abelson August 6, 1998 17:15

REVISED5 REPORT ON THE ALGORITHMIC LANGUAGE SCHEME 45

(exact? z) procedure
(inexact? z) procedure

These numerical predicates provide tests for the exactness of a quantity. For any Scheme
number, precisely one of these predicates is true.

(= z1 z2 z3 . . .) procedure
(< x1 x2 x3 . . .) procedure
(> x1 x2 x3 . . .) procedure
(<= x1 x2 x3 . . .) procedure
(>= x1 x2 x3 . . .) procedure

These procedures return#t if their arguments are (respectively): equal, monotonically
increasing, monotonically decreasing, monotonically nondecreasing, or monotonically non-
increasing.

These predicates are required to be transitive.

Note: The traditional implementations of these predicates in Lisp-like languages are not
transitive.

Note: While it is not an error to compare inexact numbers using these predicates, the
results may be unreliable because a small inaccuracy may affect the result; this is especially
true of= andzero?. When in doubt, consult a numerical analyst.

(zero? z) library procedure
(positive? x) library procedure
(negative? x) library procedure
(odd? n) library procedure
(even? n) library procedure

These numerical predicates test a number for a particular property, returning#t or #f.
See note above.

(max x1 x2 . . .) library procedure
(min x1 x2 . . .) library procedure

These procedures return the maximum or minimum of their arguments.

(max 3 4) H⇒ 4 ; exact
(max 3.9 4) H⇒ 4.0 ; inexact

Note: If any argument is inexact, then the result will also be inexact (unless the procedure
can prove that the inaccuracy is not large enough to affect the result, which is possible
only in unusual implementations). Ifmin or max is used to compare numbers of mixed
exactness, and the numerical value of the result cannot be represented as an inexact number

without loss of accuracy, then the procedure may report a violation of an implementation
restriction.

(+ z1 . . .) procedure
(* z1 . . .) procedure

These procedures return the sum or product of their arguments.

(+ 3 4) H⇒ 7
(+ 3) H⇒ 3
(+) H⇒ 0
(* 4) H⇒ 4
(*) H⇒ 1

(- z1 z2) procedure
(- z) procedure
(- z1 z2 . . .) optional procedure
(/ z1 z2) procedure
(/ z) procedure
(/ z1 z2 . . .) optional procedure

With two or more arguments, these procedures return the difference or quotient of their
arguments, associating to the left. With one argument, however, they return the additive or
multiplicative inverse of their argument.

(- 3 4) H⇒ -1
(- 3 4 5) H⇒ -6
(- 3) H⇒ -3
(/ 3 4 5) H⇒ 3/20
(/ 3) H⇒ 1/3

(abs x) library procedure

Abs returns the absolute value of its argument.

(abs -7) H⇒ 7

(quotient n1 n2) procedure
(remainder n1 n2) procedure
(modulo n1 n2) procedure

These procedures implement number-theoretic (integer) division.n2 should be nonzero.
All three procedures return integers. Ifn1/n2 is an integer:

(quotient n1 n2) H⇒ n1/n2

(remainder n1 n2) H⇒ 0
(modulo n1 n2) H⇒ 0

If n1/n2 is not an integer:

(quotient n1 n2) H⇒ nq

(remainder n1 n2) H⇒ nr

(modulo n1 n2) H⇒ nm

wherenq is n1/n2 rounded towards zero, 0< |nr | < |n2|, 0 < |nm| < |n2|, nr andnm

differ from n1 by a multiple ofn2, nr has the same sign asn1, andnm has the same sign as
n2.

From this we can conclude that for integersn1 andn2 with n2 not equal to 0,

(= n1 (+ (* n2 (quotient n1 n2))
(remainder n1 n2))) H⇒ #t

provided all numbers involved in that computation are exact.

(modulo 13 4) H⇒ 1
(remainder 13 4) H⇒ 1

(modulo -13 4) H⇒ 3
(remainder -13 4) H⇒ -1

(modulo 13 -4) H⇒ -3
(remainder 13 -4) H⇒ 1

(modulo -13 -4) H⇒ -1
(remainder -13 -4) H⇒ -1

(remainder -13 -4.0) H⇒ -1.0 ; inexact

(gcd n1 . . .) library procedure
(lcm n1 . . .) library procedure

These procedures return the greatest common divisor or least common multiple of their
arguments. The result is always nonnegative.

(gcd 32 -36) H⇒ 4
(gcd) H⇒ 0
(lcm 32 -36) H⇒ 288
(lcm 32.0 -36) H⇒ 288.0 ; inexact
(lcm) H⇒ 1

(numerator q) procedure
(denominator q) procedure

These procedures return the numerator or denominator of their argument; the result is
computed as if the argument was represented as a fraction in lowest terms. The denominator
is always positive. The denominator of 0 is defined to be 1.

(numerator (/ 6 4)) H⇒ 3
(denominator (/ 6 4)) H⇒ 2
(denominator
(exact->inexact (/ 6 4))) H⇒ 2.0

(floor x) procedure
(ceiling x) procedure
(truncate x) procedure
(round x) procedure

These procedures return integers.Floor returns the largest integer not larger thanx.
Ceiling returns the smallest integer not smaller thanx. Truncate returns the integer
closest tox whose absolute value is not larger than the absolute value ofx. Round returns
the closest integer tox, rounding to even whenx is halfway between two integers.

Rationale:Round rounds to even for consistency with the default rounding mode specified
by the IEEE floating point standard.

Note: If the argument to one of these procedures is inexact, then the result will also be
inexact. If an exact value is needed, the result should be passed to theinexact->exact
procedure.

(floor -4.3) H⇒ -5.0
(ceiling -4.3) H⇒ -4.0
(truncate -4.3) H⇒ -4.0
(round -4.3) H⇒ -4.0

(floor 3.5) H⇒ 3.0
(ceiling 3.5) H⇒ 4.0
(truncate 3.5) H⇒ 3.0
(round 3.5) H⇒ 4.0 ; inexact

(round 7/2) H⇒ 4 ; exact
(round 7) H⇒ 7

(rationalize x y) library procedure

Rationalize returns thesimplestrational number differing fromx by no more thany. A
rational numberr1 is simplerthan another rational numberr2 if r1 = p1/q1 andr2 = p2/q2

(in lowest terms) and|p1| ≤ |p2| and|q1| ≤ |q2|. Thus 3/5 is simpler than 4/7. Although
not all rationals are comparable in this ordering (consider 2/7 and 3/5) any interval contains
a rational number that is simpler than every other rational number in that interval (the simpler
2/5 lies between 2/7 and 3/5). Note that 0= 0/1 is the simplest rational of all.

(rationalize
(inexact->exact .3) 1/10) H⇒ 1/3 ; exact

(rationalize .3 1/10) H⇒ #i1/3 ; inexact

(exp z) procedure
(log z) procedure
(sin z) procedure
(cos z) procedure
(tan z) procedure
(asin z) procedure
(acos z) procedure
(atan z) procedure
(atan y x) procedure

These procedures are part of every implementation that supports general real numbers;
they compute the usual transcendental functions.Log computes the natural logarithm of
z (not the base ten logarithm).Asin, acos, andatan compute arcsine (sin−1), arccosine
(cos−1), and arctangent (tan−1), respectively. The two-argument variant ofatan computes
(angle (make-rectangular x y)) (see below), even in implementations that do not
support general complex numbers.

In general, the mathematical functions log, arcsine, arccosine, and arctangent are multiply
defined. The value of logz is defined to be the one whose imaginary part lies in the range
from −π (exclusive) toπ (inclusive). log 0 is undefined. With log defined this way, the
values of sin−1 z, cos−1 z, and tan−1 z are according to the following formulæ:

sin−1 z = −i log(i z+
√

1− z2)

cos−1 z = π/2− sin−1 z

tan−1 z = (log(1+ i z)− log(1− i z))/(2i )

The above specification follows [27], which in turn cites [19]; refer to these sources for
more detailed discussion of branch cuts, boundary conditions, and implementation of these
functions. When it is possible these procedures produce a real result from a real argument.

(sqrt z) procedure

Returns the principal square root ofz. The result will have either positive real part, or
zero real part and nonnegative imaginary part.

(expt z1 z2) procedure

Returnsz1 raised to the powerz2. Forz1 6= 0

z1
z2 = ez2 logz1

0z is 1 if z= 0 and 0 otherwise.

(make-rectangular x1 x2) procedure
(make-polar x3 x4) procedure
(real-part z) procedure
(imag-part z) procedure
(magnitude z) procedure
(angle z) procedure

These procedures are part of every implementation that supports general complex num-
bers. Supposex1, x2, x3, andx4 are real numbers andz is a complex number such that

z= x1+ x2i = x3 · eix4

Then

(make-rectangular x1 x2) H⇒ z
(make-polar x3 x4) H⇒ z
(real-part z) H⇒ x1

(imag-part z) H⇒ x2

(magnitude z) H⇒ |x3|
(angle z) H⇒ xangle

where−π < xangle≤ π with xangle= x4+ 2πn for some integern.

Rationale: Magnitude is the same asabs for a real argument, butabs must be present
in all implementations, whereasmagnitude need only be present in implementations that
support general complex numbers.

(exact->inexact z) procedure
(inexact->exact z) procedure

Exact->inexact returns an inexact representation ofz. The value returned is the inexact
number that is numerically closest to the argument. If an exact argument has no reasonably
close inexact equivalent, then a violation of an implementation restriction may be reported.
Inexact->exact returns an exact representation ofz. The value returned is the exact

number that is numerically closest to the argument. If an inexact argument has no reasonably
close exact equivalent, then a violation of an implementation restriction may be reported.

These procedures implement the natural one-to-one correspondence between exact and
inexact integers throughout an implementation-dependent range. See Section 7.2.3.

7.2.6. Numerical input and output.

(number->string z) procedure
(number->string z radix) procedure

Radixmust be an exact integer, either 2, 8, 10, or 16. If omitted,radix defaults to 10. The
procedurenumber->string takes a number and a radix and returns as a string an external
representation of the given number in the given radix such that

(let ((number number)
(radix radix))

(eqv? number
(string->number (number->string number

radix)
radix)))

is true. It is an error if no possible result makes this expression true.
If z is inexact, the radix is 10, and the above expression can be satisfied by a result that

contains a decimal point, then the result contains a decimal point and is expressed using the
minimum number of digits (exclusive of exponent and trailing zeroes) needed to make the
above expression true [3, 5]; otherwise the format of the result is unspecified.

The result returned bynumber->string never contains an explicit radix prefix.

Note: The error case can occur only whenz is not a complex number or is a complex
number with a non-rational real or imaginary part.

Rationale: If z is an inexact number represented using flonums, and the radix is 10, then
the above expression is normally satisfied by a result containing a decimal point. The
unspecified case allows for infinities, NaNs, and nonflonum representations.

(string->number string) procedure
(string->number string radix) procedure

Returns a number of the maximally precise representation expressed by the givenstring.
Radixmust be an exact integer, either 2, 8, 10, or 16. If supplied,radix is a default radix
that may be overridden by an explicit radix prefix instring (e.g.,"#o177"). If radix is not
supplied, then the default radix is 10. Ifstring is not a syntactically valid notation for a
number, thenstring->number returns#f.

(string->number "100") H⇒ 100
(string->number "100" 16) H⇒ 256
(string->number "1e2") H⇒ 100.0
(string->number "15##") H⇒ 1500.0

Note: The domain ofstring->numbermay be restricted by implementations in the follow-
ing ways.String->number is permitted to return#f wheneverstringcontains an explicit

radix prefix. If all numbers supported by an implementation are real, thenstring->number
is permitted to return#fwheneverstringuses the polar or rectangular notations for complex
numbers. If all numbers are integers, thenstring->number may return#f whenever the
fractional notation is used. If all numbers are exact, thenstring->number may return
#f whenever an exponent marker or explicit exactness prefix is used, or if a# appears in
place of a digit. If all inexact numbers are integers, thenstring->number may return#f
whenever a decimal point is used.

7.3. Other data types

This section describes operations on some of Scheme’s non-numeric data types: booleans,
pairs, lists, symbols, characters, strings and vectors.

7.3.1. Booleans. The standard boolean objects for true and false are written as#t and
#f. What really matters, though, are the objects that the Scheme conditional expressions
(if, cond, and, or, do) treat as true or false. The phrase “a true value” (or sometimes just
“true”) means any object treated as true by the conditional expressions, and the phrase “a
false value” (or “false”) means any object treated as false by the conditional expressions.

Of all the standard Scheme values, only#f counts as false in conditional expressions.
Except for#f, all standard Scheme values, including#t, pairs, the empty list, symbols,
numbers, strings, vectors, and procedures, count as true.

Note: Programmers accustomed to other dialects of Lisp should be aware that Scheme
distinguishes both#f and the empty list from the symbolnil.

Boolean constants evaluate to themselves, so they do not need to be quoted in programs.

#t H⇒ #t
#f H⇒ #f
’#f H⇒ #f

(not obj) library procedure

Not returns#t if obj is false, and returns#f otherwise.

(not #t) H⇒ #f
(not 3) H⇒ #f
(not (list 3)) H⇒ #f
(not #f) H⇒ #t
(not ’()) H⇒ #f
(not (list)) H⇒ #f
(not ’nil) H⇒ #f

(boolean? obj) library procedure

Boolean? returns#t if obj is either#t or #f and returns#f otherwise.

(boolean? #f) H⇒ #t
(boolean? 0) H⇒ #f
(boolean? ’()) H⇒ #f

7.3.2. Pairs and lists. A pair (sometimes called adotted pair) is a record structure with
two fields called the car and cdr fields (for historical reasons). Pairs are created by the
procedurecons. The car and cdr fields are accessed by the procedurescar andcdr. The
car and cdr fields are assigned by the proceduresset-car! andset-cdr!.

Pairs are used primarily to represent lists. A list can be defined recursively as either the
empty list or a pair whose cdr is a list. More precisely, the set of lists is defined as the
smallest setX such that

• The empty list is inX.
• If list is in X, then any pair whose cdr field containslist is also inX.

The objects in the car fields of successive pairs of a list are the elements of the list. For
example, a two-element list is a pair whose car is the first element and whose cdr is a pair
whose car is the second element and whose cdr is the empty list. The length of a list is the
number of elements, which is the same as the number of pairs.

The empty list is a special object of its own type (it is not a pair); it has no elements and
its length is zero.

Note: The above definitions imply that all lists have finite length and are terminated by the
empty list.

The most general notation (external representation) for Scheme pairs is the “dotted” nota-
tion (c1 . c2) wherec1 is the value of the car field andc2 is the value of the cdr field. For
example(4 . 5) is a pair whose car is 4 and whose cdr is 5. Note that(4 . 5) is the external
representation of a pair, not an expression that evaluates to a pair.

A more streamlined notation can be used for lists: the elements of the list are simply
enclosed in parentheses and separated by spaces. The empty list is written(). For example,

(a b c d e)

and

(a . (b . (c . (d . (e . ())))))

are equivalent notations for a list of symbols.
A chain of pairs not ending in the empty list is called animproper list. Note that an

improper list is not a list. The list and dotted notations can be combined to represent
improper lists:

(a b c . d)

is equivalent to

(a . (b . (c . d)))

Whether a given pair is a list depends upon what is stored in the cdr field. When the
set-cdr! procedure is used, an object can be a list one moment and not the next:

(define x (list ’a ’b ’c))
(define y x)
y H⇒ (a b c)
(list? y) H⇒ #t
(set-cdr! x 4) H⇒ unspecified
x H⇒ (a . 4)
(eqv? x y) H⇒ #t
y H⇒ (a . 4)
(list? y) H⇒ #f
(set-cdr! x x) H⇒ unspecified
(list? x) H⇒ #f

Within literal expressions and representations of objects read by theread procedure,
the forms’〈datum〉, `〈datum〉, ,〈datum〉, and,@〈datum〉 denote two-element lists whose
first elements are the symbolsquote, quasiquote, unquote, andunquote-splicing,
respectively. The second element in each case is〈datum〉. This convention is supported so
that arbitrary Scheme programs may be represented as lists. That is, according to Scheme’s
grammar, every〈expression〉 is also a〈datum〉 (see Section 8.1.2). Among other things,
this permits the use of theread procedure to parse Scheme programs. See Section 4.3.

(pair? obj) procedure

Pair? returns#t if obj is a pair, and otherwise returns#f.

(pair? ’(a . b)) H⇒ #t
(pair? ’(a b c)) H⇒ #t
(pair? ’()) H⇒ #f
(pair? ’#(a b)) H⇒ #f

(cons obj1 obj2) procedure

Returns a newly allocated pair whose car isobj1 and whose cdr isobj2. The pair is
guaranteed to be different (in the sense ofeqv?) from every existing object.

(cons ’a ’()) H⇒ (a)
(cons ’(a) ’(b c d)) H⇒ ((a) b c d)
(cons "a" ’(b c)) H⇒ ("a" b c)
(cons ’a 3) H⇒ (a . 3)
(cons ’(a b) ’c) H⇒ ((a b) . c)

(car pair) procedure

Returns the contents of the car field ofpair. Note that it is an error to take the car of the
empty list.

(car ’(a b c)) H⇒ a
(car ’((a) b c d)) H⇒ (a)
(car ’(1 . 2)) H⇒ 1
(car ’()) H⇒ error

(cdr pair) procedure

Returns the contents of the cdr field ofpair. Note that it is an error to take the cdr of the
empty list.

(cdr ’((a) b c d)) H⇒ (b c d)
(cdr ’(1 . 2)) H⇒ 2
(cdr ’()) H⇒ error

(set-car! pair obj) procedure

Storesobj in the car field ofpair. The value returned byset-car! is unspecified.

(define (f) (list ’not-a-constant-list))
(define (g) ’(constant-list))
(set-car! (f) 3) H⇒ unspecified
(set-car! (g) 3) H⇒ error

(set-cdr! pair obj) procedure

Storesobj in the cdr field ofpair. The value returned byset-cdr! is unspecified.

(caar pair) library procedure
(cadr pair) library procedure

...
...

(cdddar pair) library procedure
(cddddr pair) library procedure

These procedures are compositions ofcar andcdr, where for examplecaddr could be
defined by

(define caddr (lambda (x) (car (cdr (cdr x))))).

Arbitrary compositions, up to four deep, are provided. There are twenty-eight of these
procedures in all.

(null? obj) library procedure

Returns#t if obj is the empty list, otherwise returns#f.

(list? obj) library procedure

Returns#t if obj is a list, otherwise returns#f. By definition, all lists have finite length
and are terminated by the empty list.

(list? ’(a b c)) H⇒ #t
(list? ’()) H⇒ #t
(list? ’(a . b)) H⇒ #f
(let ((x (list ’a)))
(set-cdr! x x)
(list? x)) H⇒ #f

(list obj . . .) library procedure

Returns a newly allocated list of its arguments.

(list ’a (+ 3 4) ’c) H⇒ (a 7 c)
(list) H⇒ ()

(length list) library procedure

Returns the length oflist.

(length ’(a b c)) H⇒ 3
(length ’(a (b) (c d e))) H⇒ 3
(length ’()) H⇒ 0

(append list . . .) library procedure

Returns a list consisting of the elements of the firstlist followed by the elements of the
otherlists.

(append ’(x) ’(y)) H⇒ (x y)
(append ’(a) ’(b c d)) H⇒ (a b c d)
(append ’(a (b)) ’((c))) H⇒ (a (b) (c))

The resulting list is always newly allocated, except that it shares structure with the last
list argument. The last argument may actually be any object; an improper list results if the
last argument is not a proper list.

(append ’(a b) ’(c . d)) H⇒ (a b c . d)
(append ’() ’a) H⇒ a

(reverse list) library procedure

Returns a newly allocated list consisting of the elements oflist in reverse order.

(reverse ’(a b c)) H⇒ (c b a)
(reverse ’(a (b c) d (e (f)))) H⇒ ((e (f)) d (b c) a)

(list-tail list k) library procedure

Returns the sublist oflist obtained by omitting the firstk elements. It is an error iflist
has fewer thank elements.List-tail could be defined by

(define list-tail
(lambda (x k)
(if (zero? k)

x
(list-tail (cdr x) (- k 1)))))

(list-ref list k) library procedure

Returns thekth element oflist. (This is the same as the car of(list-tail list k).) It
is an error iflist has fewer thank elements.

(list-ref ’(a b c d) 2) H⇒ c
(list-ref ’(a b c d)

(inexact->exact (round 1.8)))
H⇒ c

(memq obj list) library procedure
(memv obj list) library procedure
(member obj list) library procedure

These procedures return the first sublist oflist whose car isobj, where the sublists of
list are the non-empty lists returned by(list-tail list k) for k less than the length of
list. If obj does not occur inlist, then#f (not the empty list) is returned.Memq useseq? to
compareobj with the elements oflist, while memv useseqv? andmember usesequal?.

(memq ’a ’(a b c)) H⇒ (a b c)
(memq ’b ’(a b c)) H⇒ (b c)
(memq ’a ’(b c d)) H⇒ #f

(memq (list ’a) ’(b (a) c)) H⇒ #f
(member (list ’a)

’(b (a) c)) H⇒ ((a) c)
(memq 101 ’(100 101 102)) H⇒ unspecified
(memv 101 ’(100 101 102)) H⇒ (101 102)

(assq obj alist) library procedure
(assv obj alist) library procedure
(assoc obj alist) library procedure

Alist (for “association list”) must be a list of pairs. These procedures find the first pair in
alist whose car field isobj, and returns that pair. If no pair inalist hasobj as its car, then
#f (not the empty list) is returned.Assq useseq? to compareobj with the car fields of the
pairs inalist, while assv useseqv? andassoc usesequal?.

(define e ’((a 1)(b 2)(c 3)))
(assq ’a e) H⇒ (a 1)
(assq ’b e) H⇒ (b 2)
(assq ’d e) H⇒ #f
(assq (list ’a) ’(((a)) ((b)) ((c))))

H⇒ #f
(assoc (list ’a) ’(((a)) ((b)) ((c))))

H⇒ ((a))
(assq 5 ’((2 3) (5 7) (11 13))) H⇒ unspecified
(assv 5 ’((2 3) (5 7) (11 13))) H⇒ (5 7)

Rationale: Although they are ordinarily used as predicates,memq, memv, member, assq,
assv, andassoc do not have question marks in their names because they return useful
values rather than just#t or #f.

7.3.3. Symbols. Symbols are objects whose usefulness rests on the fact that two symbols
are identical (in the sense ofeqv?) if and only if their names are spelled the same way.
This is exactly the property needed to represent identifiers in programs, and so most imple-
mentations of Scheme use them internally for that purpose. Symbols are useful for many
other applications; for instance, they may be used the way enumerated values are used in
Pascal.

The rules for writing a symbol are exactly the same as the rules for writing an identifier;
see Sections 3.1 and 8.1.1.

It is guaranteed that any symbol that has been returned as part of a literal expression, or
read using theread procedure, and subsequently written out using thewrite procedure,
will read back in as the identical symbol (in the sense ofeqv?). Thestring->symbol
procedure, however, can create symbols for which this write/read invariance may not hold
because their names contain special characters or letters in the nonstandard case.

Note: Some implementations of Scheme have a feature known as “slashification” in order
to guarantee write/read invariance for all symbols, but historically the most important use
of this feature has been to compensate for the lack of a string data type.

Some implementations also have “uninterned symbols”, which defeat write/read invari-
ance even in implementations with slashification, and also generate exceptions to the rule
that two symbols are the same if and only if their names are spelled the same.

(symbol? obj) procedure

Returns#t if obj is a symbol, otherwise returns#f.

(symbol? ’foo) H⇒ #t
(symbol? (car ’(a b))) H⇒ #t
(symbol? "bar") H⇒ #f
(symbol? ’nil) H⇒ #t
(symbol? ’()) H⇒ #f
(symbol? #f) H⇒ #f

(symbol->string symbol) procedure

Returns the name ofsymbolas a string. If the symbol was part of an object returned
as the value of a literal expression (Section 5.1.2) or by a call to theread procedure, and
its name contains alphabetic characters, then the string returned will contain characters
in the implementation’s preferred standard case—some implementations will prefer upper
case, others lower case. If the symbol was returned bystring->symbol, the case of
characters in the string returned will be the same as the case in the string that was passed to
string->symbol. It is an error to apply mutation procedures likestring-set! to strings
returned by this procedure.

The following examples assume that the implementation’s standard case is lower case:

(symbol->string ’flying-fish) H⇒ "flying-fish"
(symbol->string ’Martin) H⇒ "martin"
(symbol->string
(string->symbol "Malvina")) H⇒ "Malvina"

(string->symbol string) procedure

Returns the symbol whose name isstring. This procedure can create symbols with names
containing special characters or letters in the nonstandard case, but it is usually a bad idea
to create such symbols because in some implementations of Scheme they cannot be read as
themselves. Seesymbol->string.

The following examples assume that the implementation’s standard case is lower case:

(eq? ’mISSISSIppi ’mississippi) H⇒ #t
(string->symbol "mISSISSIppi") H⇒ the symbol with name

"mISSISSIppi"

(eq? ’bitBlt (string->symbol "bitBlt"))
H⇒ #f

(eq? ’JollyWog
(string->symbol
(symbol->string ’JollyWog)))

H⇒ #t
(string=? "K. Harper, M.D."

(symbol->string
(string->symbol "K. Harper, M.D.")))

H⇒ #t

7.3.4. Characters. Characters are objects that represent printed characters such as letters
and digits. Characters are written using the notation#\〈character〉 or #\〈character name〉.
For example:

#\a ; lower case letter
#\A ; upper case letter
#\( ; left parenthesis
#\ ; the space character
#\space ; the preferred way to write a space
#\newline ; the newline character

Case is significant in#\〈character〉, but not in#\〈character name〉. If 〈character〉 in
#\〈character〉 is alphabetic, then the character following〈character〉must be a delimiter
character such as a space or parenthesis. This rule resolves the ambiguous case where, for
example, the sequence of characters “#\space” could be taken to be either a representation
of the space character or a representation of the character “#\s” followed by a representation
of the symbol “pace.”

Characters written in the#\ notation are self-evaluating. That is, they do not have to be
quoted in programs.

Some of the procedures that operate on characters ignore the difference between upper
case and lower case. The procedures that ignore case have “-ci” (for “case insensitive”)
embedded in their names.

(char? obj) procedure

Returns#t if obj is a character, otherwise returns#f.

(char=? char1 char2) procedure
(char<? char1 char2) procedure
(char>? char1 char2) procedure
(char<=? char1 char2) procedure
(char>=? char1 char2) procedure

These procedures impose a total ordering on the set of characters. It is guaranteed that
under this ordering:

• The upper case characters are in order. For example,(char<? #\A #\B) returns#t.
• The lower case characters are in order. For example,(char<? #\a #\b) returns#t.
• The digits are in order. For example,(char<? #\0 #\9) returns#t.
• Either all the digits precede all the upper case letters, or vice versa.
• Either all the digits precede all the lower case letters, or vice versa.

Some implementations may generalize these procedures to take more than two arguments,
as with the corresponding numerical predicates.

(char-ci=? char1 char2) library procedure
(char-ci<? char1 char2) library procedure
(char-ci>? char1 char2) library procedure
(char-ci<=? char1 char2) library procedure
(char-ci>=? char1 char2) library procedure

These procedures are similar tochar=? et cetera, but they treat upper case and lower
case letters as the same. For example,(char-ci=? #\A #\a) returns#t. Some imple-
mentations may generalize these procedures to take more than two arguments, as with the
corresponding numerical predicates.

(char-alphabetic? char) library procedure
(char-numeric? char) library procedure
(char-whitespace? char) library procedure
(char-upper-case? letter) library procedure
(char-lower-case? letter) library procedure

These procedures return#t if their arguments are alphabetic, numeric, whitespace, up-
per case, or lower case characters, respectively, otherwise they return#f. The following
remarks, which are specific to the ASCII character set, are intended only as a guide: The
alphabetic characters are the 52 upper and lower case letters. The numeric characters are
the ten decimal digits. The whitespace characters are space, tab, line feed, form feed, and
carriage return.

(char->integer char) procedure
(integer->char n) procedure

Given a character,char->integer returns an exact integer representation of the char-
acter. Given an exact integer that is the image of a character underchar->integer,
integer->char returns that character. These procedures implement order-preserving iso-
morphisms between the set of characters under thechar<=? ordering and some subset of
the integers under the<= ordering. That is, if

(char<=? a b H⇒ #t and (<= x y) H⇒ #t

andx andy are in the domain ofinteger->char, then

(<= (char->integer a)
(char->integer b)) H⇒ #t

(char<=? (integer->char x)
(integer->char y)) H⇒ #t

(char-upcase char) library procedure
(char-downcase char) library procedure

These procedures return a characterchar2 such that(char-ci=? char char2). In ad-
dition, if char is alphabetic, then the result ofchar-upcase is upper case and the result of
char-downcase is lower case.

7.3.5. Strings. Strings are sequences of characters. Strings are written as sequences of
characters enclosed within doublequotes ("). A doublequote can be written inside a string
only by escaping it with a backslash (\), as in

"The word \"recursion\" has many meanings."

A backslash can be written inside a string only by escaping it with another backslash.
Scheme does not specify the effect of a backslash within a string that is not followed by a
doublequote or backslash.

A string constant may continue from one line to the next, but the exact contents of such
a string are unspecified.

The lengthof a string is the number of characters that it contains. This number is an
exact, non-negative integer that is fixed when the string is created. Thevalid indexesof a
string are the exact nonnegative integers less than the length of the string. The first character
of a string has index 0, the second has index 1, and so on.

In phrases such as “the characters ofstring beginning with indexstart and ending with
indexend,” it is understood that the indexstart is inclusive and the indexendis exclusive.
Thus if start andendare the same index, a null substring is referred to, and ifstart is zero
andendis the length ofstring, then the entire string is referred to.

Some of the procedures that operate on strings ignore the difference between upper and
lower case. The versions that ignore case have “-ci” (for “case insensitive”) embedded in
their names.

(string? obj) procedure

Returns#t if obj is a string, otherwise returns#f.

(make-string k) procedure
(make-string k char) procedure

Make-string returns a newly allocated string of lengthk. If char is given, then all
elements of the string are initialized tochar, otherwise the contents of thestring are un-
specified.

(string char . . .) library procedure

Returns a newly allocated string composed of the arguments.

(string-length string) procedure

Returns the number of characters in the givenstring.

(string-ref string k) procedure

k must be a valid index ofstring. String-ref returns characterk of stringusing zero-origin
indexing.

(string-set! string k char) procedure

k must be a valid index ofstring. String-set! storeschar in elementk of string and
returns an unspecified value.

(define (f) (make-string 3 #\*))
(define (g) "***")
(string-set! (f) 0 #\?) H⇒ unspecified
(string-set! (g) 0 #\?) H⇒ error
(string-set! (symbol->string ’immutable)

0
#\?) H⇒ error

(string=? string1 string2) library procedure
(string-ci=? string1 string2) library procedure

Returns#t if the two strings are the same length and contain the same characters in the
same positions, otherwise returns#f. String-ci=? treats upper and lower case letters as
though they were the same character, butstring=? treats upper and lower case as distinct
characters.

(string<? string1 string2) library procedure
(string>? string1 string2) library procedure
(string<=? string1 string2) library procedure
(string>=? string1 string2) library procedure
(string-ci<? string1 string2) library procedure
(string-ci>? string1 string2) library procedure
(string-ci<=? string1 string2) library procedure
(string-ci>=? string1 string2) library procedure

These procedures are the lexicographic extensions to strings of the corresponding order-
ings on characters. For example,string<? is the lexicographic ordering on strings induced

by the orderingchar<? on characters. If two strings differ in length but are the same up to
the length of the shorter string, the shorter string is considered to be lexicographically less
than the longer string.

Implementations may generalize these and thestring=? andstring-ci=? procedures
to take more than two arguments, as with the corresponding numerical predicates.

(substring string start end) library procedure

Stringmust be a string, andstart andendmust be exact integers satisfying

0 ≤ start≤ end≤ (string-length string).

Substring returns a newly allocated string formed from the characters ofstringbeginning
with indexstart (inclusive) and ending with indexend(exclusive).

(string-append string . . . ) library procedure

Returns a newly allocated string whose characters form the concatenation of the given
strings.

(string->list string) library procedure
(list->string list) library procedure

String->list returns a newly allocated list of the characters that make up the given
string. List->string returns a newly allocated string formed from the characters in the
list list, which must be a list of characters.String->list andlist->string are inverses
so far asequal? is concerned.

(string-copy string) library procedure

Returns a newly allocated copy of the givenstring.

(string-fill! string char) library procedure

Storeschar in every element of the givenstringand returns an unspecified value.

7.3.6. Vectors. Vectors are heterogenous structures whose elements are indexed by inte-
gers. A vector typically occupies less space than a list of the same length, and the average
time required to access a randomly chosen element is typically less for the vector than for
the list.

The length of a vector is the number of elements that it contains. This number is a
nonnegative integer that is fixed when the vector is created. Thevalid indexesof a vector
are the exact nonnegative integers less than the length of the vector. The first element in a
vector is indexed by zero, and the last element is indexed by one less than the length of the
vector.

Vectors are written using the notation#(obj . . .). For example, a vector of length 3
containing the number zero in element 0, the list(2 2 2 2) in element 1, and the string
"Anna" in element 2 can be written as following:

#(0 (2 2 2 2) "Anna")

Note that this is the external representation of a vector, not an expression evaluating to a
vector. Like list constants, vector constants must be quoted:

’#(0 (2 2 2 2) "Anna") H⇒ #(0 (2 2 2 2) "Anna")

(vector? obj) procedure

Returns#t if obj is a vector, otherwise returns#f.

(make-vector k) procedure
(make-vector k fill) procedure

Returns a newly allocated vector ofk elements. If a second argument is given, then each
element is initialized tofill . Otherwise the initial contents of each element is unspecified.

(vector obj . . .) library procedure

Returns a newly allocated vector whose elements contain the given arguments. Analogous
to list.

(vector ’a ’b ’c) H⇒ #(a b c)

(vector-length vector) procedure

Returns the number of elements invectoras an exact integer.

(vector-ref vector k) procedure

k must be a valid index ofvector. Vector-ref returns the contents of elementk of vector.

(vector-ref ’#(1 1 2 3 5 8 13 21)
5) H⇒ 8

(vector-ref ’#(1 1 2 3 5 8 13 21)
(let ((i (round (* 2 (acos -1)))))
(if (inexact? i)
(inexact->exact i)
i))) H⇒ 13

(vector-set! vector k obj) procedure

k must be a valid index ofvector. Vector-set! storesobj in elementk of vector. The
value returned byvector-set! is unspecified.

(let ((vec (vector 0 ’(2 2 2 2) "Anna")))
(vector-set! vec 1 ’("Sue" "Sue"))
vec) H⇒ #(0 ("Sue" "Sue") "Anna")

(vector-set! ’#(0 1 2) 1 "doe") H⇒ error ; constant vector

(vector->list vector) library procedure
(list->vector list) library procedure

Vector->list returns a newly allocated list of the objects contained in the elements of
vector. List->vector returns a newly created vector initialized to the elements of the list
list.

(vector->list ’#(dah dah didah)) H⇒ (dah dah didah)
(list->vector ’(dididit dah)) H⇒ #(dididit dah)

(vector-fill! vector fill) library procedure

Storesfill in every element ofvector. The value returned byvector-fill! is unspeci-
fied.

7.4. Control features

This section describes various primitive procedures which control the flow of program
execution in special ways. Theprocedure? predicate is also described here.

(procedure? obj) procedure

Returns#t if obj is a procedure, otherwise returns#f.

(procedure? car) H⇒ #t
(procedure? ’car) H⇒ #f
(procedure? (lambda (x) (* x x)))

H⇒ #t
(procedure? ’(lambda (x) (* x x)))

H⇒ #f
(call-with-current-continuation procedure?)

H⇒ #t

(apply proc arg1 . . . args) procedure

Proc must be a procedure andargsmust be a list. Callsproc with the elements of the list
(append (list arg1 . . .) args) as the actual arguments.

(apply + (list 3 4)) H⇒ 7

(define compose
(lambda (f g)
(lambda args
(f (apply g args)))))

((compose sqrt *) 12 75) H⇒ 30

(map proc list1 list2 . . .) library procedure

The lists must be lists, andprocmust be a procedure taking as many arguments as there
arelists and returning a single value. If more than onelist is given, then they must all be
the same length.Map appliesproc element-wise to the elements of thelists and returns a
list of the results, in order. The dynamic order in whichproc is applied to the elements of
the lists is unspecified.

(map cadr ’((a b) (d e) (g h))) H⇒ (b e h)

(map (lambda (n) (expt n n))
’(1 2 3 4 5)) H⇒ (1 4 27 256 3125)

(map + ’(1 2 3) ’(4 5 6)) H⇒ (5 7 9)

(let ((count 0))
(map (lambda (ignored)

(set! count (+ count 1))
count)

’(a b))) H⇒ (1 2) or (2 1)

(for-each proc list1 list2 . . .) library procedure

The arguments tofor-each are like the arguments tomap, but for-each calls proc
for its side effects rather than for its values. Unlikemap, for-each is guaranteed to call
procon the elements of thelists in order from the first element(s) to the last, and the value
returned byfor-each is unspecified.

(let ((v (make-vector 5)))
(for-each (lambda (i)

(vector-set! v i (* i i)))
’(0 1 2 3 4))

v) H⇒ #(0 1 4 9 16)

(force promise) library procedure

Forces the value ofpromise(seedelay, Section 5.2.5). If no value has been computed
for the promise, then a value is computed and returned. The value of the promise is cached
(or “memoized”) so that if it is forced a second time, the previously computed value is
returned.

(force (delay (+ 1 2))) H⇒ 3
(let ((p (delay (+ 1 2))))
(list (force p) (force p))) H⇒ (3 3)

(define a-stream
(letrec ((next

(lambda (n)
(cons n (delay (next (+ n 1)))))))

(next 0)))
(define head car)
(define tail
(lambda (stream) (force (cdr stream))))

(head (tail (tail a-stream))) H⇒ 2

Force anddelay are mainly intended for programs written in functional style. The
following examples should not be considered to illustrate good programming style, but
they illustrate the property that only one value is computed for a promise, no matter how
many times it is forced.

(define count 0)
(define p
(delay (begin (set! count (+ count 1))

(if (> count x)
count
(force p)))))

(define x 5)
p H⇒ a promise
(force p) H⇒ 6
p H⇒ a promise, still
(begin (set! x 10)

(force p)) H⇒ 6

Here is a possible implementation ofdelay andforce. Promises are implemented here
as procedures of no arguments, andforce simply calls its argument:

(define force
(lambda (object)
(object)))

We define the expression

(delay 〈expression〉)

to have the same meaning as the procedure call

(make-promise (lambda () 〈expression〉))

as follows

(define-syntax delay
(syntax-rules ()
((delay expression)
(make-promise (lambda () expression))))),

wheremake-promise is defined as follows:

(define make-promise
(lambda (proc)
(let ((result-ready? #f)

(result #f))
(lambda ()
(if result-ready?

result
(let ((x (proc)))
(if result-ready?

result
(begin (set! result-ready? #t))

(set! result x)
result))))))))

Rationale: A promise may refer to its own value, as in the last example above. Forcing
such a promise may cause the promise to be forced a second time before the value of the
first force has been computed. This complicates the definition ofmake-promise.

Various extensions to this semantics ofdelay andforce are supported in some imple-
mentations:

• Callingforce on an object that is not a promise may simply return the object.

• It may be the case that there is no means by which a promise can be operationally
distinguished from its forced value. That is, expressions like the following may evaluate
to either#t or to#f, depending on the implementation:

(eqv? (delay 1) 1) H⇒ unspecified
(pair? (delay (cons 1 2))) H⇒ unspecified

• Some implementations may implement “implicit forcing,” where the value of a promise
is forced by primitive procedures likecdr and+:

(+ (delay (* 3 7)) 13) H⇒ 34

(call-with-current-continuation proc) procedure

Proc must be a procedure of one argument. The procedurecall-with-current-
continuation packages up the current continuation (see the rationale below) as an “es-
cape procedure” and passes it as an argument toproc. The escape procedure is a Scheme
procedure that, if it is later called, will abandon whatever continuation is in effect at that
later time and will instead use the continuation that was in effect when the escape procedure
was created. Calling the escape procedure may cause the invocation ofbeforeandafter
thunks installed usingdynamic-wind.

The escape procedure accepts the same number of arguments as the continuation to the
original call tocall-with-current-continuation. Except for continuations created
by thecall-with-values procedure, all continuations take exactly one value. The effect
of passing no value or more than one value to continuations that were not created by
call-with-values is unspecified.

The escape procedure that is passed toproc has unlimited extent just like any other
procedure in Scheme. It may be stored in variables or data structures and may be called as
many times as desired.

The following examples show only the most common ways in whichcall-with-
current-continuation is used. If all real uses were as simple as these examples, there
would be no need for a procedure with the power ofcall-with-current-
continuation.

(call-with-current-continuation
(lambda (exit)
(for-each (lambda (x)

(if (negative? x)
(exit x)))

’(54 0 37 -3 245 19))
#t)) H⇒ -3

(define list-length

(lambda (obj)
(call-with-current-continuation
(lambda (return)
(letrec ((r

(lambda (obj)
(cond ((null? obj) 0)

((pair? obj)
(+ (r (cdr obj)) 1))

(else (return #f))))))
(r obj))))))

(list-length ’(1 2 3 4)) H⇒ 4

(list-length ’(a b . c)) H⇒ #f

Rationale:A common use ofcall-with-current-continuation is for structured, non-
local exits from loops or procedure bodies, but in factcall-with-current-
continuation is extremely useful for implementing a wide variety of advanced con-
trol structures.

Whenever a Scheme expression is evaluated there is acontinuationwanting the result of
the expression. The continuation represents an entire (default) future for the computation.
If the expression is evaluated at top level, for example, then the continuation might take the
result, print it on the screen, prompt for the next input, evaluate it, and so on forever. Most
of the time the continuation includes actions specified by user code, as in a continuation
that will take the result, multiply it by the value stored in a local variable, add seven,
and give the answer to the top level continuation to be printed. Normally these ubiquitous
continuations are hidden behind the scenes and programmers do not think much about them.
On rare occasions, however, a programmer may need to deal with continuations explicitly.
Call-with-current-continuation allows Scheme programmers to do that by creating
a procedure that acts just like the current continuation.

Most programming languages incorporate one or more special-purpose escape constructs
with names likeexit, return, or evengoto. In 1965, however, Peter Landin [16] invented
a general purpose escape operator called the J-operator. John Reynolds [24] described a sim-
pler but equally powerful construct in 1972. Thecatch special form described by Sussman
and Steele in the 1975 report on Scheme is exactly the same as Reynolds’s construct, though
its name came from a less general construct in MacLisp. Several Scheme implementors
noticed that the full power of thecatch construct could be provided by a procedure instead
of by a special syntactic construct, and the namecall-with-current-continuation
was coined in 1982. This name is descriptive, but opinions differ on the merits of such a
long name, and some people use the namecall/cc instead.

(values obj . . .) procedure

Delivers all of its arguments to its continuation. Except for continuations created by the
call-with-values procedure, all continuations take exactly one value.

Values might be defined as follows:

(define (values . things)
(call-with-current-continuation
(lambda (cont) (apply cont things))))

(call-with-values producer consumer) procedure

Calls itsproducerargument with no values and a continuation that, when passed some
values, calls theconsumerprocedure with those values as arguments. The continuation for
the call toconsumeris the continuation of the call tocall-with-values.

(call-with-values (lambda () (values 4 5))
(lambda (a b) b))

H⇒ 5

(call-with-values * -) H⇒ -1

(dynamic-wind before thunk after) procedure

Calls thunkwithout arguments, returning the result(s) of this call.Beforeandafter are
called, also without arguments, as required by the following rules (note that in the absence
of calls to continuations captured usingcall-with-current-continuation the three
arguments are called once each, in order).Beforeis called whenever execution enters the
dynamic extent of the call tothunkandafter is called whenever it exits that dynamic extent.
The dynamic extent of a procedure call is the period between when the call is initiated
and when it returns. In Scheme, because ofcall-with-current-continuation, the
dynamic extent of a call may not be a single, connected time period. It is defined as
follows:

• The dynamic extent is entered when execution of the body of the called procedure begins.
• The dynamic extent is also entered when execution is not within the dynamic ex-

tent and a continuation is invoked that was captured (usingcall-with-current-
continuation) during the dynamic extent.
• It is exited when the called procedure returns.
• It is also exited when execution is within the dynamic extent and a continuation is invoked

that was captured while not within the dynamic extent.

If a second call todynamic-wind occurs within the dynamic extent of the call tothunk
and then a continuation is invoked in such a way that theafters from these two invocations
of dynamic-wind are both to be called, then theafter associated with the second (inner)
call todynamic-wind is called first.

If a second call todynamic-wind occurs within the dynamic extent of the call tothunk
and then a continuation is invoked in such a way that thebefores from these two invocations

of dynamic-wind are both to be called, then thebeforeassociated with the first (outer) call
to dynamic-wind is called first.

If invoking a continuation requires calling thebeforefrom one call todynamic-wind
and theafter from another, then theafter is called first.

The effect of using a captured continuation to enter or exit the dynamic extent of a call
to beforeor after is undefined.

(let ((path ’())
(c #f))

(let ((add (lambda (s)
(set! path (cons s path)))))

(dynamic-wind
(lambda () (add ’connect))
(lambda ()
(add (call-with-current-continuation

(lambda (c0)
(set! c c0)
’talk1))))

(lambda () (add ’disconnect)))
(if (< (length path) 4)

(c ’talk2)
(reverse path)))) H⇒ (connect talk1 disconnect

connect talk2 disconnect)

7.5. Eval

(eval expression environment-specifier) procedure

Evaluatesexpressionin the specified environment and returns its value.Expressionmust
be a valid Scheme expression represented as data, andenvironment-specifiermust be a value
returned by one of the three procedures described below. Implementations may extendeval
to allow nonexpression programs (definitions) as the first argument and to allow other values
as environments, with the restriction thateval is not allowed to create new bindings in the
environments associated withnull-environment or scheme-report-environment.

(eval ’(* 7 3) (scheme-report-environment 5))
H⇒ 21

(let ((f (eval ’(lambda (f x) (f x x))
(null-environment 5))))

(f + 10)) H⇒ 20

(scheme-report-environment version) procedure
(null-environment version) procedure

Versionmust be the exact integer5, corresponding to this revision of the Scheme report
(the Revised5 Report on Scheme).Scheme-report-environment returns a specifier for an
environment that is empty except for all bindings defined in this report that are either required
or both optional and supported by the implementation.Null-environment returns a
specifier for an environment that is empty except for the (syntactic) bindings for all syntactic
keywords defined in this report that are either required or both optional and supported by
the implementation.

Other values ofversioncan be used to specify environments matching past revisions
of this report, but their support is not required. An implementation will signal an error if
versionis neither5 nor another value supported by the implementation.

The effect of assigning (through the use ofeval) a variable bound in ascheme-report-
environment (for examplecar) is unspecified. Thus the environments specified by
scheme-report-environment may be immutable.

(interaction-environment) optional procedure

This procedure returns a specifier for the environment that contains implementation-
defined bindings, typically a superset of those listed in the report. The intent is that this
procedure will return the environment in which the implementation would evaluate expres-
sions dynamically typed by the user.

7.6. Input and output

7.6.1. Ports. Ports represent input and output devices. To Scheme, an input port is a
Scheme object that can deliver characters upon command, while an output port is a Scheme
object that can accept characters.

(call-with-input-file string proc) library procedure
(call-with-output-file string proc) library procedure

Stringshould be a string naming a file, andproc should be a procedure that accepts one
argument. Forcall-with-input-file, the file should already exist; forcall-with-
output-file, the effect is unspecified if the file already exists. These procedures call
proc with one argument: the port obtained by opening the named file for input or output.
If the file cannot be opened, an error is signalled. Ifproc returns, then the port is closed
automatically and the value(s) yielded by theproc is(are) returned. Ifprocdoes not return,
then the port will not be closed automatically unless it is possible to prove that the port will
never again be used for a read or write operation.

Rationale: Because Scheme’s escape procedures have unlimited extent, it is possible to
escape from the current continuation but later to escape back in. If implementations were
permitted to close the port on any escape from the current continuation, then it would be
impossible to write portable code using bothcall-with-current-continuation and
call-with-input-file or call-with-output-file.

(input-port? obj) procedure
(output-port? obj) procedure

Returns#t if obj is an input port or output port respectively, otherwise returns#f.

(current-input-port) procedure
(current-output-port) procedure

Returns the current default input or output port.

(with-input-from-file string thunk) optional procedure
(with-output-to-file string thunk) optional procedure

String should be a string naming a file, andproc should be a procedure of no argu-
ments. Forwith-input-from-file, the file should already exist; forwith-output-to-
file, the effect is unspecified if the file already exists. The file is opened for input
or output, an input or output port connected to it is made the default value returned
by current-input-port or current-output-port (and is used by(read), (write
obj), and so forth), and thethunk is called with no arguments. When thethunkreturns,
the port is closed and the previous default is restored.With-input-from-file and
with-output-to-file return(s) the value(s) yielded bythunk. If an escape procedure is
used to escape from the continuation of these procedures, their behavior is implementation
dependent.

(open-input-file filename) procedure

Takes a string naming an existing file and returns an input port capable of delivering
characters from the file. If the file cannot be opened, an error is signalled.

(open-output-file filename) procedure

Takes a string naming an output file to be created and returns an output port capable
of writing characters to a new file by that name. If the file cannot be opened, an error is
signalled. If a file with the given name already exists, the effect is unspecified.

(close-input-port port) procedure
(close-output-port port) procedure

Closes the file associated withport, rendering theport incapable of delivering or accepting
characters. These routines have no effect if the file has already been closed. The value
returned is unspecified.

7.6.2. Input.

(read) library procedure
(read port) library procedure

Read converts external representations of Scheme objects into the objects themselves.
That is, it is a parser for the nonterminal〈datum〉 (see Sections 8.1.2 and 7.3.2).Read
returns the next object parsable from the given inputport, updatingport to point to the first
character past the end of the external representation of the object.

If an end of file is encountered in the input before any characters are found that can
begin an object, then an end of file object is returned. The port remains open, and further
attempts to read will also return an end of file object. If an end of file is encountered
after the beginning of an object’s external representation, but the external representation is
incomplete and therefore not parsable, an error is signalled.

The port argument may be omitted, in which case it defaults to the value returned by
current-input-port. It is an error to read from a closed port.

(read-char) procedure
(read-char port) procedure

Returns the next character available from the inputport, updating theport to point to the
following character. If no more characters are available, an end of file object is returned.Port
may be omitted, in which case it defaults to the value returned bycurrent-input-port.

(peek-char) procedure
(peek-char port) procedure

Returns the next character available from the inputport, without updating theport to
point to the following character. If no more characters are available, an end of file ob-
ject is returned.Port may be omitted, in which case it defaults to the value returned by
current-input-port.

Note: The value returned by a call topeek-char is the same as the value that would have
been returned by a call toread-char with the sameport. The only difference is that the
very next call toread-char or peek-char on thatport will return the value returned by
the preceding call topeek-char. In particular, a call topeek-char on an interactive port
will hang waiting for input whenever a call toread-char would have hung.

(eof-object? obj) procedure

Returns#t if obj is an end of file object, otherwise returns#f. The precise set of end of
file objects will vary among implementations, but in any case no end of file object will ever
be an object that can be read in usingread.

(char-ready?) procedure
(char-ready? port) procedure

Returns#t if a character is ready on the inputport and returns#t otherwise. If
char-ready returns#t then the nextread-char operation on the givenport is guar-
anteed not to hang. If theport is at end of file thenchar-ready? returns#t. Port may be
omitted, in which case it defaults to the value returned bycurrent-input-port.

Rationale:Char-ready? exists to make it possible for a program to accept characters from
interactive ports without getting stuck waiting for input. Any input editors associated with
such ports must ensure that characters whose existence has been asserted byChar-ready?
cannot be rubbed out. IfChar-ready? were to return#f at end of file, a port at end of file
would be indistinguishable from an interactive port that has no ready characters.

7.6.3. Output.

(write obj) library procedure
(write obj port) library procedure

Writes a written representation ofobj to the givenport. Strings that appear in the written
representation are enclosed in doublequotes, and within those strings backslash and dou-
blequote characters are escaped by backslashes. Character objects are written using the#\
notation.Write returns an unspecified value. Theport argument may be omitted, in which
case it defaults to the value returned bycurrent-output-port.

(display obj) library procedure
(display obj port) library procedure

Writes a representation ofobj to the givenport. Strings that appear in the written rep-
resentation are not enclosed in doublequotes, and no characters are escaped within those
strings. Character objects appear in the representation as if written bywrite-char instead
of by write. Display returns an unspecified value. Theport argument may be omitted,
in which case it defaults to the value returned bycurrent-output-port.

Rationale: Write is intended for producing machine-readable output anddisplay is
for producing human-readable output. Implementations that allow “slashification” within
symbols will probably wantwritebut notdisplay to slashify funny characters in symbols.

(newline) library procedure
(newline port) library procedure

Writes an end of line toport. Exactly how this is done differs from one operating system
to another. Returns an unspecified value. Theport argument may be omitted, in which case
it defaults to the value returned bycurrent-output-port.

(write-char char) procedure
(write-char char port) procedure

Writes the characterchar (not an external representation of the character) to the given
port and returns an unspecified value. Theport argument may be omitted, in which case it
defaults to the value returned bycurrent-output-port.

7.6.4. System interface.Questions of system interface generally fall outside of the do-
main of this report. However, the following operations are important enough to deserve
description here.

(load filename) optional procedure

Filename should be a string naming an existing file containing Scheme source code. The
load procedure reads expressions and definitions from the file and evaluates them se-
quentially. It is unspecified whether the results of the expressions are printed. Theload
procedure does not affect the values returned bycurrent-input-port andcurrent-
output-port. Load returns an unspecified value.

Rationale:For portability,load must operate on source files. Its operation on other kinds
of files necessarily varies among implementations.

(transcript-on filename) optional procedure
(transcript-off) optional procedure

Filename must be a string naming an output file to be created. The effect of
transcript-on is to open the named file for output, and to cause a transcript of sub-
sequent interaction between the user and the Scheme system to be written to the file. The
transcript is ended by a call totranscript-off, which closes the transcript file. Only
one transcript may be in progress at any time, though some implementations may relax this
restriction. The values returned by these procedures are unspecified.

8. Formal syntax and semantics

This chapter provides formal descriptions of what has already been described informally in
previous sections of this report.

8.1. Formal syntax

This section provides a formal syntax for Scheme written in an extended BNF.
All spaces in the grammar are for legibility. Case is insignificant; for example,#x1A and

#X1a are equivalent.〈empty〉 stands for the empty string.
The following extensions to BNF are used to make the description more concise:〈thing〉∗

means zero or more occurrences of〈thing〉; and〈thing〉+ means at least one〈thing〉.

8.1.1. Lexical structure. This section describes how individual tokens (identifiers, num-
bers, etc.) are formed from sequences of characters. The following sections describe how

expressions and programs are formed from sequences of tokens.〈Intertoken space〉 may
occur on either side of any token, but not within a token.

Tokens which require implicit termination (identifiers, numbers, characters, and dot) may
be terminated by any〈delimiter〉, but not necessarily by anything else.

The following five characters are reserved for future extensions to the language:[ ]
{ } |

〈token〉 −→ 〈identifier〉 | 〈boolean〉 | 〈number〉
| 〈character〉 | 〈string〉
| ( | ) | #( | ’ | ` | , | ,@ | .

〈delimiter〉 −→ 〈whitespace〉 | ( | ) | " | ;
〈whitespace〉 −→ 〈space or newline〉
〈comment〉 −→ ; 〈all subsequent characters up to a

line break〉
〈atmosphere〉 −→ 〈whitespace〉 | 〈comment〉
〈intertoken space〉 −→ 〈atmosphere〉∗

〈identifier〉 −→ 〈initial〉 〈subsequent〉∗
| 〈peculiar identifier〉

〈initial〉 −→ 〈letter〉 | 〈special initial〉
〈letter〉 −→ a | b | c | ... | z

〈special initial〉 −→ ! | $ | % | & | * | / | : | < | =
| > | ? | ^ | _ | ~

〈subsequent〉 −→ 〈initial〉 | 〈digit〉
| 〈special subsequent〉

〈digit〉 −→ 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
〈special subsequent〉 −→ + | - | . | @
〈peculiar identifier〉 −→ + | - | ...
〈syntactic keyword〉 −→ 〈expression keyword〉
| else | => | define
| unquote | unquote-splicing

〈expression keyword〉 −→ quote | lambda | if
| set! | begin | cond | and | or | case
| let | let* | letrec | do | delay
| quasiquote

〈variable〉 −→ 〈any〈identifier〉 that isn’t
also a〈syntactic keyword〉〉

〈boolean〉 −→ #t | #f
〈character〉 −→ #\ 〈any character〉
| #\ 〈character name〉

〈character name〉 −→ space | newline

〈string〉 −→ " 〈string element〉∗ "

〈string element〉 −→ 〈any character other than" or \〉
| \" | \\

〈number〉 −→ 〈num 2〉| 〈num 8〉
| 〈num 10〉| 〈num 16〉

The following rules for〈num R〉, 〈complexR〉, 〈real R〉, 〈ureal R〉, 〈uintegerR〉, and
〈prefix R〉 should be replicated forR= 2, 8, 10, and 16. There are no rules for〈decimal 2〉,
〈decimal 8〉, and〈decimal 16〉, which means that numbers containing decimal points or
exponents must be in decimal radix.

〈num R〉 −→ 〈prefix R〉 〈complexR〉
〈complexR〉 −→ 〈real R〉 | 〈real R〉 @ 〈real R〉
| 〈real R〉 + 〈urealR〉 i | 〈real R 〉 - 〈urealR〉 i
| 〈real R〉 + i | 〈real R〉 - i
| + 〈urealR〉 i | - 〈urealR〉 i | + i | - i

〈real R〉 −→ 〈sign〉 〈urealR〉
〈urealR〉 −→ 〈uintegerR〉
| 〈uintegerR〉 / 〈uintegerR〉
| 〈decimalR〉

〈decimal 10〉 −→ 〈uinteger 10〉 〈suffix〉
| . 〈digit 10〉+ #∗ 〈suffix〉
| 〈digit 10〉+ . 〈digit 10〉∗ #∗ 〈suffix〉
| 〈digit 10〉+ #+ . #∗ 〈suffix〉

〈uintegerR〉 −→ 〈digit R〉+ #∗
〈prefix R〉 −→ 〈radix R〉 〈exactness〉
| 〈exactness〉 〈radix R〉

〈suffix〉 −→ 〈empty〉
| 〈exponent marker〉 〈sign〉 〈digit 10〉+

〈exponent marker〉 −→ e | s | f | d | l
〈sign〉 −→ 〈empty〉 | + | -
〈exactness〉 −→ 〈empty〉 | #i | #e
〈radix 2〉 −→ #b
〈radix 8〉 −→ #o
〈radix 10〉 −→ 〈empty〉 | #d
〈radix 16〉 −→ #x
〈digit 2〉 −→ 0 | 1
〈digit 8〉 −→ 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7
〈digit 10〉 −→ 〈digit〉
〈digit 16〉 −→ 〈digit 10〉 | a | b | c | d | e | f

8.1.2. External representations.〈Datum〉 is what theread procedure (Section 7.6.2)
successfully parses. Note that any string that parses as an〈expression〉 will also parse as a
〈datum〉.

〈datum〉 −→ 〈simple datum〉 | 〈compound datum〉
〈simple datum〉 −→ 〈boolean〉 | 〈number〉
| 〈character〉 | 〈string〉 | 〈symbol〉

〈symbol〉 −→ 〈identifier〉
〈compound datum〉 −→ 〈list〉 | 〈vector〉
〈list〉 −→ (〈datum〉∗)| (〈datum〉+ . 〈datum〉)
| 〈abbreviation〉

〈abbreviation〉 −→ 〈abbrev prefix〉 〈datum〉
〈abbrev prefix〉 −→ ’ | ` | , | ,@
〈vector〉 −→ #(〈datum〉*)

8.1.3. Expressions.

〈expression〉 −→ 〈variable〉
| 〈literal〉
| 〈procedure call〉
| 〈lambda expression〉
| 〈conditional〉
| 〈assignment〉
| 〈derived expression〉
| 〈macro use〉
| 〈macro block〉

〈literal〉 −→ 〈quotation〉 | 〈self-evaluating〉
〈self-evaluating〉 −→ 〈boolean〉 | 〈number〉
| 〈character〉 | 〈string〉

〈quotation〉 −→ ’〈datum〉 | (quote 〈datum〉)
〈procedure call〉 −→ (〈operator〉 〈operand〉∗)
〈operator〉 −→ 〈expression〉
〈operand〉 −→ 〈expression〉
〈lambda expression〉 −→ (lambda 〈formals〉 〈body〉)
〈formals〉 −→ (〈variable〉∗)| 〈variable〉
| (〈variable〉+ . 〈variable〉)

〈body〉 −→ 〈definition〉∗ 〈sequence〉
〈sequence〉 −→ 〈command〉∗ 〈expression〉
〈command〉 −→ 〈expression〉

〈conditional〉 −→ (if 〈test〉 〈consequent〉 〈alternate〉)
〈test〉 −→ 〈expression〉
〈consequent〉 −→ 〈expression〉
〈alternate〉 −→ 〈expression〉 | 〈empty〉

〈assignment〉 −→ (set! 〈variable〉 〈expression〉)

〈derived expression〉 −→
(cond 〈cond clause〉+)
| (cond 〈cond clause〉∗ (else 〈sequence〉))
| (case 〈expression〉
〈case clause〉+)

| (case 〈expression〉
〈case clause〉∗
(else 〈sequence〉))

| (and 〈test〉∗)
| (or 〈test〉∗)
| (let (〈binding spec〉∗) 〈body〉)
| (let 〈variable〉 (〈binding spec〉∗) 〈body〉)
| (let* (〈binding spec〉∗) 〈body〉)
| (letrec (〈binding spec〉∗) 〈body〉)
| (begin 〈sequence〉)
| (do (〈iteration spec〉∗)

(〈test〉 〈do result〉)
〈command〉∗)

| (delay 〈expression〉)
| 〈quasiquotation〉

〈cond clause〉 −→ (〈test〉 〈sequence〉)
| (〈test〉)
| (〈test〉 => 〈recipient〉)

〈recipient〉 −→ 〈expression〉
〈case clause〉 −→ ((〈datum〉∗) 〈sequence〉)
〈binding spec〉 −→ (〈variable〉 〈expression〉)
〈iteration spec〉 −→ (〈variable〉 〈init〉 〈step〉)
| (〈variable〉 〈init〉)

〈init〉 −→ 〈expression〉
〈step〉 −→ 〈expression〉
〈do result〉 −→ 〈sequence〉 | 〈empty〉

〈macro use〉 −→ (〈keyword〉 〈datum〉∗)
〈keyword〉 −→ 〈identifier〉

〈macro block〉 −→
(let-syntax (〈syntax spec〉∗) 〈body〉)
| (letrec-syntax (〈syntax spec〉∗) 〈body〉)

〈syntax spec〉 −→ (〈keyword〉 〈transformer spec〉)

8.1.4. Quasiquotations. The following grammar for quasiquote expressions is not context-
free. It is presented as a recipe for generating an infinite number of production rules. Imagine
a copy of the following rules forD = 1, 2, 3, . . . . D keeps track of the nesting depth.

〈quasiquotation〉 −→ 〈quasiquotation 1〉
〈qq template 0〉 −→ 〈expression〉
〈quasiquotationD〉 −→ `〈qq templateD〉
| (quasiquote 〈qq templateD〉)

〈qq templateD〉 −→ 〈simple datum〉
| 〈list qq templateD〉
| 〈vector qq templateD〉
| 〈unquotationD〉

〈list qq templateD〉 −→ (〈qq template or spliceD〉∗)
| (〈qq template or spliceD〉+ . 〈qq templateD〉)
| ’〈qq templateD〉
| 〈quasiquotationD + 1〉

〈vector qq templateD〉 −→ #(〈qq template or spliceD〉∗)
〈unquotationD〉 −→ ,〈qq templateD − 1〉
| (unquote 〈qq templateD − 1〉)

〈qq template or spliceD〉 −→ 〈qq templateD〉
| 〈splicing unquotationD〉

〈splicing unquotationD〉 −→ ,@〈qq templateD − 1〉
| (unquote-splicing 〈qq templateD − 1〉)

In 〈quasiquotation〉s, a〈list qq templateD〉 can sometimes be confused with either an
〈unquotationD〉 or a 〈splicing unquotationD〉. The interpretation as an〈unquotation〉 or
〈splicing unquotationD〉 takes precedence.

8.1.5. Transformers.

〈transformer spec〉 −→
(syntax-rules (〈identifier〉∗) 〈syntax rule〉∗)

〈syntax rule〉 −→ (〈pattern〉 〈template〉)
〈pattern〉 −→ 〈pattern identifier〉

| (〈pattern〉∗)
| (〈pattern〉+ . 〈pattern〉)
| (〈pattern〉∗ 〈pattern〉 〈ellipsis〉)
| #(〈pattern〉∗)
| #(〈pattern〉∗ 〈pattern〉 〈ellipsis〉)
| 〈pattern datum〉

〈pattern datum〉 −→ 〈string〉
| 〈character〉
| 〈boolean〉
| 〈number〉

〈template〉 −→ 〈pattern identifier〉
| (〈template element〉∗)
| (〈template element〉+ · 〈template〉)
| #(〈template element〉∗)
| 〈template datum〉

〈template element〉 −→ 〈template〉
| 〈template〉 〈ellipsis〉

〈template datum〉 −→ 〈pattern datum〉
〈pattern identifier〉 −→ 〈any identifier except...〉
〈ellipsis〉 −→ 〈the identifier...〉

8.1.6. Programs and definitions.

〈program〉 −→ 〈command or definition〉∗
〈command or definition〉 −→ 〈command〉
| 〈definition〉
| 〈syntax definition〉
| (begin 〈command or definition〉+)

〈definition〉 −→ (define 〈variable〉 〈expression〉)
| (define (〈variable〉 〈def formals〉) 〈body〉)
| (begin 〈definition〉∗)

〈def formals〉 −→ 〈variable〉∗
| 〈variable〉∗ . 〈variable〉

〈syntax definition〉 −→
(define-syntax 〈keyword〉 〈transformer spec〉)

8.2. Formal semantics

This section provides a formal denotational semantics for the primitive expressions of
Scheme and selected built-in procedures. The concepts and notation used here are de-
scribed in [29]; the notation is summarized below:

〈 . . . 〉 sequence formation
s ↓ k kth member of the sequences (1-based)
#s length of sequences
s § t concatenation of sequencess andt
s † k drop the firstk members of sequences
t → a, b McCarthy conditional “ift thena elseb”
ρ[x/ i ] substitution “ρ with x for i ”
x in D injection ofx into domainD
x | D projection ofx to domainD

The reason that expression continuations take sequences of values instead of single values
is to simplify the formal treatment of procedure calls and multiple return values.

The boolean flag associated with pairs, vectors, and strings will be true for mutable
objects and false for immutable objects.

The order of evaluation within a call is unspecified. We mimic that here by applying
arbitrary permutationspermuteandunpermute, which must be inverses, to the arguments
in a call before and after they are evaluated. This is not quite right since it suggests,

incorrectly, that the order of evaluation is constant throughout a program (for any given
number of arguments), but it is a closer approximation to the intended semantics than a
left-to-right evaluation would be.

The storage allocatornewis implementation-dependent, but it must obey the following
axiom: if newσ ∈ L, thenσ(newσ | L) ↓ 2= false.

The definition ofK is omitted because an accurate definition ofK would complicate the
semantics without being very interesting.

If P is a program in which all variables are defined before being referenced or assigned,
then the meaning of P is

E [[((lambda (I*) P’) 〈undefined〉 . . . )]]

where I∗ is the sequence of variables defined in P, P′ is the sequence of expressions obtained
by replacing every definition in P by an assignment,〈undefined〉 is an expression that
evaluates toundefined, andE is the semantic function that assigns meaning to expressions.

8.2.1. Abstract syntax.

K ∈Con constants, including quotations
I ∈ Ide identifiers (variables)
E∈Exp expressions
0 ∈Com=Exp commands

Exp−→K | I | (E0 E∗)
| (lambda (I∗) 0∗ E0)
| (lambda (I∗ . I) 0∗ E0)
| (lambda I 0∗ E0)
| (if E0 E1 E2) | (if E0 E1)
| (set! I E)

8.2.2. Domain equations.

α ∈ L locations
ν ∈ N natural numbers

T = { false, true} booleans
Q symbols
H characters
R numbers
Ep = L× L× T pairs
Ev = L∗ × T vectors
Es = L∗ × T strings
M = { false, true, null, undefined, unspecified}

miscellaneous
φ ∈ F = L× (E∗ → K→ C) procedure values
ε ∈ E = Q+ H+ R+ Ep+ Ev + Es+ M+ F

expressed values

σ ∈ S = L→ (E× T) stores
ρ ∈ U = Ide→ L environments
θ ∈ C = S→ A command continuations
κ ∈ K = E∗ → C expression continuations

A answers
X errors

8.2.3. Semantic functions.

K : Con→ E
E : Exp→ U→ K→ C
E∗ : Exp∗ → U→ K→ C
C : Com∗ → U→ C→ C

Definition ofK deliberately omitted.

E [[K]] = λρκ . send(K[[K]] ) κ

E [[I]] = λρκ . hold(lookupρ I)
(single(λε . ε = undefined→

wrong“undefined variable”,
sendε κ))

E [[(E0 E∗)]] =
λρκ . E∗(permute(〈E0〉 § E∗))

ρ

(λε∗ . ((λε∗ . applicate(ε∗ ↓ 1) (ε∗ † 1) κ)
(unpermuteε∗)))

E [[(lambda (I∗) 0∗ E0)]] =
λρκ . λσ .

newσ ∈ L→
send(〈newσ | L,

λε∗κ ′ . #ε∗ = #I∗ →
tievals(λα∗ . (λρ ′ . C[[0∗]]ρ ′(E [[E0]]ρ ′κ ′))

(extendsρ I∗ α∗))
ε∗,

wrong“wrong number of arguments”〉
in E)

κ

(update(newσ | L) unspecifiedσ),
wrong“out of memory”σ

E [[(lambda (I∗ . I) 0∗ E0)]] =
λρκ . λσ .

newσ ∈ L→
send(〈newσ | L,

λε∗κ ′ . #ε∗ ≥ #I→
tievalsrest
(λα∗ . (λρ ′ . C[[0∗]]ρ ′(E [[E0]]ρ ′κ ′))

(extendsρ (I∗ § 〈I〉) α∗))
ε∗

(#I∗),
wrong“too few arguments”〉 in E)

κ

(update(newσ | L) unspecifiedσ),
wrong“out of memory”σ

E [[(lambda I 0∗ E0)]] = E [[(lambda (. I) 0∗ E0)]]

E [[(if E0 E1 E2)]] =
λρκ . E [[E0]] ρ (single(λε . truishε → E [[E1]]ρκ,

E [[E2]]ρκ))

E [[(if E0 E1)]] =
λρκ . E [[E0]] ρ (single(λε . truishε → E [[E1]]ρκ,

send unspecifiedκ))

Here and elsewhere, any expressed value other thanundefinedmay be used in place of
unspecified.

E [[(set! I E)]] =
λρκ . E [E] ρ (single(λε . assign(lookupρ I)

ε

(send unspecifiedκ)))

E∗[[ ]] = λρκ . κ〈 〉

E∗[[E0 E∗]] =
λρκ . E [[E0]] ρ (single(λε0 . E∗[[E∗]] ρ (λε∗ . κ (〈ε0〉 §ε∗))))

C[[ ]] = λρθ . θ

C[[000
∗]] = λρθ . E [[00]] ρ (λε∗ . C[[0∗]]ρθ)

8.2.4. Auxiliary functions.

lookup: U→ Ide→ L
lookup= λρI . ρI

extends: U→ Ide∗ → L∗ → U
extends=
λρI∗α∗ . #I∗ = 0→ ρ,

extends(ρ[(α∗ ↓ 1)/(I∗ ↓ 1)]) (I∗ † 1) (α∗ † 1)

wrong : X→ C [implementation-dependent]

send: E→ K→ C
send= λεκ . κ〈ε〉

single: (E→ C)→ K
single=
λψε∗ . #ε∗ = 1→ ψ(ε∗ ↓ 1),

wrong“wrong number of return values”

new: S→ (L+ {error}) [implementation-dependent]

hold : L→ K→ C
hold= λακσ . send(σα ↓ 1)κσ

assign: L→ E→ C→ C
assign= λαεθσ . θ(updateαεσ)

update: L→ E→ S→ S
update= λαεσ . σ [〈ε, true〉/α]

tievals: (L∗ → C)→ E∗ → C
tievals=
λψε∗σ . #ε∗ = 0→ ψ〈 〉σ,

newσ ∈ L→ tievals(λα∗ . ψ(〈newσ | L〉 §α∗))
(ε∗ † 1)
(update(newσ | L)(ε∗ ↓ 1)σ ),

wrong“out of memory”σ

tievalsrest: (L∗ → C)→ E∗ → N→ C
tievalsrest=
λψε∗ν . list (dropfirstε∗ν)

(single(λε . tievalsψ ((takefirstε∗ν) § 〈ε〉)))

dropfirst= λln . n = 0→ l , dropfirst(l † 1)(n− 1)

takefirst= λln . n = 0→ 〈 〉, 〈l ↓ 1〉 §(takefirst(l † 1)(n− 1))

truish : E→ T
truish= λε . ε = false→ false, true



P1: VBI

HIGHER-ORDER AND SYMBOLIC COMPUTATION KL610-01-Abelson August 6, 1998 17:15

REVISED5 REPORT ON THE ALGORITHMIC LANGUAGE SCHEME 89

permute: Exp∗ → Exp∗ [implementation-dependent]

unpermute: E∗ → E∗ [inverse ofpermute]

applicate: E→ E∗ → K→ C
applicate=
λεε∗κ . ε ∈ F→ (ε | F ↓ 2)ε∗κ,wrong“bad procedure”

onearg: (E→ K→ C)→ (E∗ → K→ C)
onearg=
λζε∗κ . #ε∗ = 1→ ζ(ε∗ ↓ 1)κ,

wrong“wrong number of arguments”

twoarg : (E→ E→ K→ C)→ (E∗ → K→ C)
twoarg=
λζε∗κ . #ε∗ = 2→ ζ(ε∗ ↓ 1)(ε∗ ↓ 2)κ,

wrong“wrong number of arguments”

list : E∗ → K→ C
list =
λε∗κ . #ε∗ = 0→ send nullκ,

list (ε∗ † 1)(single(λε . cons〈ε∗ ↓ 1, ε〉κ))

cons: E∗ → K→ C
cons=

twoarg(λε1ε2κσ . newσ ∈ L→
(λσ ′ . newσ ′ ∈ L→

send(〈newσ | L, newσ ′ | L, true〉
in E)
κ

(update(newσ ′ | L)ε2σ
′),

wrong“out of memory”σ ′)
(update(newσ | L)ε1σ),

wrong“out of memory”σ)

less: E∗ → K→ C
less=

twoarg(λε1ε2κ . (ε1∈ R ∧ ε2∈ R)→
send(ε1 | R < ε2 | R→ true, false)κ,
wrong“non-numeric argument to<”)

add : E∗ → K→ C
add=

twoarg(λε1ε2κ . (ε1∈ R ∧ ε2∈ R)→
send((ε1 | R+ ε2 | R) in E)κ,
wrong“non-numeric argument to+”)

car : E∗ → K→ C
car =

onearg(λεκ . ε ∈ Ep→ hold(ε | Ep ↓ 1)κ,
wrong“non-pair argument tocar”)

cdr : E∗ → K→ C [similar tocar]

setcar: E∗ → K→ C
setcar=

twoarg(λε1ε2κ . ε1∈ Ep→
(ε1 | Ep ↓ 3)→ assign(ε1 | Ep ↓ 1)

ε2

(send unspecifiedκ),
wrong“immutable argument toset-car!” ,
wrong“non-pair argument toset-car!”)

eqv: E∗ → K→ C
eqv=

twoarg(λε1ε2κ . (ε1∈ M ∧ ε2∈ M)→
send(ε1 | M = ε2 | M→ true, false)κ,

(ε1∈ Q ∧ ε2∈ Q)→
send(ε1 | Q = ε2 | Q→ true, false)κ,

(ε1∈ H ∧ ε2∈ H)→
send(ε1 | H = ε2 | H→ true, false)κ,

(ε1∈ R ∧ ε2∈ R)→
send(ε1 | R = ε2 | R→ true, false)κ,

(ε1∈ Ep ∧ ε2∈ Ep)→
send((λp1 p2 . ((p1 ↓ 1) = (p2 ↓ 1)∧

(p1 ↓ 2) = (p2 ↓ 2))→ true,
false)

(ε1 | Ep)

(ε2 | Ep))

κ,

(ε1∈ Ev ∧ ε2∈ Ev)→ . . . ,

(ε1∈ Es∧ ε2∈ Es)→ . . . ,

(ε1∈ F ∧ ε2∈ F)→
send((ε1 | F ↓ 1) = (ε2 | F ↓ 1)→ true, false)

κ,

send falseκ)

apply : E∗ → K→ C
apply=

twoarg(λε1ε2κ . ε1∈ F→ valueslist〈ε2〉(λε∗ . applicateε1ε
∗κ),

wrong“bad procedure argument toapply”)

valueslist: E∗ → K→ C

valueslist=
onearg(λεκ . ε ∈ Ep→

cdr〈ε〉
(λε∗ . valueslist

ε∗

(λε∗ . car〈ε〉(single(λε . κ(〈ε〉 §ε∗))))),
ε = null→ κ〈 〉,

wrong“non-list argument tovalues-list”)

cwcc: E∗ → K→ C [call-with-current-continuation]
cwcc=

onearg(λεκ . ε ∈ F→
(λσ . newσ ∈ L→

applicateε
〈〈newσ | L, λε∗κ ′ . κε∗〉 in E〉
κ

(update(newσ | L)
unspecified
σ),

wrong“out of memory”σ),
wrong“bad procedure argument”)

values: E∗ → K→ C
values= λε∗κ . κε∗

cwv : E∗ → K→ C [call-with-values]
cwv=

twoarg(λε1ε2κ . applicateε1〈 〉(λε∗ . applicateε2 ε
∗))

8.3. Derived expression types

This section gives macro definitions for the derived expression types in terms of the primitive
expression types (literal, variable, call,lambda, if, set!). See Section 7.4 for a possible
definition ofdelay.

(define-syntax cond
(syntax-rules (else =>)
((cond (else result1 result2 ...))
(begin result1 result2 ...))
((cond (test => result))
(let ((temp test))
(if temp (result temp))))

((cond (test => result) clause1 clause2 ...)
(let ((temp test))
(if temp

(result temp)
(cond clause1 clause2 ...))))

((cond (test)) test)
((cond (test) clause1 clause2 ...)
(let ((temp test))
(if temp

temp
(cond clause1 clause2 ...))))

((cond (test result1 result2 ...))
(if test (begin result1 result2 ...)))
((cond (test result1 result2 ...)

clause1 clause2 ...)
(if test

(begin result1 result2 ...)
(cond clause1 clause2 ...)))))

(define-syntax case
(syntax-rules (else)
((case (key ...)

clauses ...)
(let ((atom-key (key ...)))
(case atom-key clauses ...)))

((case key
(else result1 result2 ...))

(begin result1 result2 ...))
((case key

((atoms ...) result1 result2 ...))
(if (memv key ’(atoms ...))
(begin result1 result2 ...)))

((case key
((atoms ...) result1 result2 ...)
clause clauses ...)

(if (memv key ’(atoms ...))
(begin result1 result2 ...)
(case key clause clauses ...)))))

(define-syntax and
(syntax-rules ()
((and) #t)
((and test) test)
((and test1 test2 ...)
(if test1 (and test2 ...) #f))))

(define-syntax or
(syntax-rules ()

((or) #f)
((or test) test)
((or test1 test2 ...)
(let ((x test1))
(if x x (or test2 ...))))))

(define-syntax let
(syntax-rules ()
((let ((name val) ...) body1 body2 ...)
((lambda (name ...) body1 body2 ...)
val ...))

((let tag ((name val) ...) body1 body2 ...)
((letrec ((tag (lambda (name ...)

body1 body2 ...)))
tag)

val ...))))

(define-syntax let*
(syntax-rules ()
((let* () body1 body2 ...)
(let () body1 body2 ...))
((let* ((name1 val1) (name2 val2) ...)

body1 body2 ...)
(let ((name1 val1))

(let* ((name2 val2) ...)
body1 body2 ...)))))

The followingletrec macro uses the symbol<undefined> in place of an expression
which returns something that when stored in a location makes it an error to try to obtain the
value stored in the location (no such expression is defined in Scheme). A trick is used to
generate the temporary names needed to avoid specifying the order in which the values are
evaluated. This could also be accomplished by using an auxiliary macro.

(define-syntax letrec
(syntax-rules ()
((letrec ((var1 init1) ...) body ...)
(letrec "generate_temp_names"
(var1 ...)
()
((var1 init1) ...)
body ...))

((letrec "generate_temp_names"
()
(temp1 ...)

((var1 init1) ...)
body ...)

(let ((var1 <undefined>) ...)
(let ((temp1 init1) ...)
(set! var1 temp1)
...
body ...)))

((letrec "generate_temp_names"
(x y ...)
(temp ...)
((var1 init1) ...)
body ...)

(letrec "generate_temp_names"
(y ...)
(newtemp temp ...)
((var1 init1) ...)
body ...))))

(define-syntax begin
(syntax-rules ()
((begin exp ...)
((lambda () exp ...)))))

The following alternative expansion forbegin does not make use of the ability to write
more than one expression in the body of a lambda expression. In any case, note that these
rules apply only if the body of thebegin contains no definitions.

(define-syntax begin
(syntax-rules ()
((begin exp)
exp)
((begin exp1 exp2 ...)
(let ((x exp1))
(begin exp2 ...)))))

The following definition ofdouses a trick to expand the variable clauses. As withletrec
above, an auxiliary macro would also work. The expression(if #f #f) is used to obtain
an unspecific value.

(define-syntax do
(syntax-rules ()
((do ((var init step ...) ...)

(test expr ...)
command ...)

(letrec
((loop
(lambda (var ...)
(if test

(begin
(if #f #f)
expr ...)

(begin
command
...
(loop (do "step" var step ...)

...))))))
(loop init ...)))

((do "step" x)
x)
((do "step" x y)
y)))

Notes

Language changes.This section enumerates the changes that have been made to Scheme
since the “Revised4 report” [6] was published.

• The report is now a superset of the IEEE standard for Scheme [13]: implementations
that conform to the report will also conform to the standard. This required the following
changes:

— The empty list is now required to count as true.
— The classification of features as essential or inessential has been removed. There

are now three classes of built-in procedures: primitive, library, and optional. The
optional procedures areload, with-input-from-file, with-output-to-file,
transcript-on, transcript-off, andinteraction-environment, and- and
/ with more than two arguments. None of these are in the IEEE standard.

— Programs are allowed to redefine built-in procedures. Doing so will not change the
behavior of other built-in procedures.

• Port has been added to the list of disjoint types.
• The macro appendix has been removed. High-level macros are now part of the main body

of the report. The rewrite rules for derived expressions have been replaced with macro
definitions. There are no reserved identifiers.
• Syntax-rules now allows vector patterns.
• Multiple-value returns,eval, anddynamic-wind have been added.
• The calls that are required to be implemented in a properly tail-recursive fashion are

defined explicitly.
• ‘@’ can be used within identifiers. ‘|’ is reserved for possible future extensions.

Additional material

The Internet Scheme Repository at

http://www.cs.indiana.edu/scheme-repository/

contains an extensive Scheme bibliography, as well as papers, programs, implementations,
and other material related to Scheme.

Example

Integrate-system integrates the system

y′k = fk(y1, y2, . . . , yn), k = 1, . . . ,n

of differential equations with the method of Runge-Kutta.
The parametersystem-derivative is a function that takes a system state (a vector

of values for the state variablesy1, . . . , yn) and produces a system derivative (the values
y′1, . . . , y′n). The parameterinitial-state provides an initial system state, andh is an
initial guess for the length of the integration step.

The value returned byintegrate-system is an infinite stream of system states.

(define integrate-system
(lambda (system-derivative initial-state h)
(let ((next (runge-kutta-4 system-derivative h)))
(letrec ((states

(cons initial-state
(delay (map-streams next

states)))))
states))))

Runge-Kutta-4 takes a function,f, that produces a system derivative from a system
state.Runge-Kutta-4 produces a function that takes a system state and produces a new
system state.

(define runge-kutta-4
(lambda (f h)
(let ((*h (scale-vector h))

(*2 (scale-vector 2))
(*1/2 (scale-vector (/ 1 2)))
(*1/6 (scale-vector (/ 1 6))))

(lambda (y)
;; y is a system state
(let* ((k0 (*h (f y)))

(k1 (*h (f (add-vectors y (*1/2 k0)))))
(k2 (*h (f (add-vectors y (*1/2 k1)))))
(k3 (*h (f (add-vectors y k2)))))

(add-vectors y
(*1/6 (add-vectors k0

(*2 k1)
(*2 k2)
k3))))))))

(define elementwise
(lambda (f)
(lambda vectors
(generate-vector
(vector-length (car vectors))
(lambda (i)
(apply f

(map (lambda (v) (vector-ref v i))
vectors)))))))

(define generate-vector
(lambda (size proc)
(let ((ans (make-vector size)))
(letrec ((loop

(lambda (i)
(cond ((= i size) ans)

(else
(vector-set! ans i (proc i))
(loop (+ i 1)))))))

(loop 0)))))
(define add-vectors (elementwise +))

(define scale-vector
(lambda (s)
(elementwise (lambda (x) (* x s)))))

Map-streams is analogous tomap: it applies its first argument (a procedure) to all the
elements of its second argument (a stream).

(define map-streams
(lambda (f s)
(cons (f (head s))

(delay (map-streams f (tail s))))))

Infinite streams are implemented as pairs whose car holds the first element of the stream
and whose cdr holds a promise to deliver the rest of the stream.

(define head car)
(define tail
(lambda (stream) (force (cdr stream))))

The following illustrates the use ofintegrate-system in integrating the system

C
dvC

dt
= −i L − vC

R

L
diL
dt
= vC

which models a damped oscillator.

(define damped-oscillator
(lambda (R L C)
(lambda (state)
(let ((Vc (vector-ref state 0))

(Il (vector-ref state 1)))
(vector (- 0 (+ (/ Vc (* R C)) (/ Il C)))

(/ Vc L))))))

(define the-states
(integrate-system

(damped-oscillator 10000 1000 .001)
’#(1 0)
.01))

Alphabetic index of definitions of concepts, keywords, and procedures

The principal entry for each term, procedure, or keyword is listed first, separated from the
other entries by a semicolon.

! 12
’ 19; 54
* 46
+ 46; 12, 89
, 28; 54
,@ 28
- 46; 12
-> 12
. . . 12; 31
/ 46
; 13
< 45; 89
<= 45
= 45
=> 22; 23
> 45
>= 45
? 12
‘ 28

abs 46; 50
acos 49
and 24; 92
angle 50
append 56
apply 67; 18, 90
asin 49
assoc 58
assq 58
assv 58
atan 49

#b 43; 80
backquote 28
begin 26; 34, 35, 94
binding 14
binding construct 14
boolean? 52; 14
bound 14
caar 55

cadr 55
call 20
call by need 28
call-with-current-continuation

70; 18, 72, 91
call-with-input-file 74
call-with-output-file 74
call-with-values 72; 18, 71, 91
call/cc 71
car 55; 90
case 23; 92
catch 71
cdddar 55
cddddr 55
cdr 55
ceiling 48
char->integer 61
char-alphabetic? 61
char-ci<=? 61
char-ci<? 61
char-ci=? 61
char-ci>=? 61
char-ci>? 61
char-downcase 62
char-lower-case? 61
char-numeric? 61
char-ready? 76
char-upcase 62
char-upper-case? 61
char-whitespace? 61
char<=? 60; 62
char<? 60
char=? 60
char>=? 60
char>? 60
char? 60; 15
close-input-port 75
close-output-port 75
combination 20
comma 28

comment 13; 79
complex? 44; 41
cond 22; 33, 91
cons 54
constant 16
continuation 71
cos 49
current-input-port 75
current-output-port 75

#d 43
define 34; 30
define-syntax 35
definition 34
delay 28; 68
denominator 48
display 77
do 26; 94
dotted pair 53
dynamic-wind 72; 70

#e 43; 80
else 22; 23
empty list 53; 15, 52, 55, 56
eof-object? 76
eq? 39; 21
equal? 40
equivalence predicate 36
eqv? 37; 16, 21, 90
error 10
escape procedure 70
eval 73; 18
even? 45
exact 37
exact->inexact 50
exact? 45
exactness 41
exp 49
expt 50

#f 52
false 15; 52
floor 48
for-each 67
force 68; 28

gcd 47

hygienic 30

#i 43; 80
identifier 12; 14, 58, 79
if 21; 87
imag-part 50
immutable 16
implementation restriction

10; 42
improper list 53
inexact 37
inexact->exact 50; 42
inexact? 45
initial environment 35
input-port? 75
integer->char 61
integer? 44; 41
interaction-environment 74
internal definition 35

keyword 29; 79

lambda 20; 35, 87
lazy evaluation 28
lcm 47
length 56; 42
let 24; 27, 33, 35, 93
let* 25; 35, 93
let-syntax 30; 35
letrec 25; 35, 93
letrec-syntax 30; 35
library 10
library procedure 36
list 56
list->string 64
list->vector 66
list-ref 57
list-tail 57
list? 56
load 78
location 15
log 49

macro 29
macro keyword 29
macro transformer 29
macro use 29
magnitude 50
make-polar 50
make-rectangular 50
make-string 62
make-vector 65
map 67
max 45
member 57
memq 57
memv 57
min 45
modulo 46
mutable 16

negative? 45
newline 77
nil 52
not 52
null-environment 73
null? 56
number 40
number->string 51
number? 44; 14, 41, 44
numerator 48
numerical types 41

#o 43; 80
object 8
odd? 45
open-input-file 75
open-output-file 75
optional 9
or 24; 93
output-port? 75

pair 53
pair? 54; 14
peek-char 76
port 74
port? 15
positive? 45

predicate 36
procedure call 20
procedure? 66; 15

promise 28; 68
proper tail recursion 16

quasiquote 28; 29, 54
quote 19; 54
quotient 46

rational? 44; 41
rationalize 48
read 76; 54, 80
read-char 76
real-part 50
real? 44; 41, 44
referentially transparent 30
region 14; 22, 24, 25, 27
remainder 46
reverse 57
round 48

scheme-report-environment 74
set! 22; 34, 87
set-car! 55
set-cdr! 55; 54
setcar 90
simplest rational 49
sin 49
sqrt 49
string 63
string->list 64
string->number 51
string->symbol 59
string-append 64
string-ci<=? 63
string-ci<? 63
string-ci=? 63
string-ci>=? 63
string-ci>? 63
string-copy 64
string-fill! 64
string-length 63; 42
string-ref 63

string-set! 63; 59
string<=? 63
string<? 63
string=? 63
string>=? 63
string>? 63
string? 62; 15
substring 64
symbol->string 59; 16
symbol? 59; 14
syntactic keyword 14; 12, 29, 79
syntax definition 35
syntax-rules 31; 36

#t 52
tail call 17
tan 49
token 79
top level environment 36; 14
transcript-off 78
transcript-on 78
true 15; 22, 23, 52
truncate 48
type 14

unbound 14; 18, 34
unquote 29; 54
unquote-splicing 29; 54
unspecified 10

valid indexes 62; 64
values 71; 20
variable 14; 12, 18, 79
vector 65
vector->list 66
vector-fill! 66
vector-length 65; 42
vector-ref 65
vector-set! 66
vector? 65; 15

whitespace 13
with-input-from-file 75
with-output-to-file 75
write 77; 29
write-char 77

#x 43; 80

zero? 45



P1: VBI

HIGHER-ORDER AND SYMBOLIC COMPUTATION KL610-01-Abelson August 6, 1998 17:15

REVISED5 REPORT ON THE ALGORITHMIC LANGUAGE SCHEME 105

Clarifications and corrections

Several inconsistencies in this report have been pointed out since its initial publication on
the World-Wide Web. This section addresses the most important of them.

Semantics ofcall-with-current-continuation

The definition ofcall-with- current-continuation in Section 8.2 is incorrect be-
cause it is incompatible withdynamic-wind. As shown in Section 4 of [1], however, this
incorrect semantics is adequate to define theshift andreset operators, which can then
be used to define the correct semantics of bothdynamic-wind andcall-with-current-
continuation.

Redefinition of keywords

As stated in Section 3.1, keywords are not reserved. The grammar specified in Section 8.1
is thus somewhat misleading because a program may redefine any or all syntactic keywords.
In particular, it is possible to use any identifier as a variable.

Port?

Section 4.2 includesport? in the list of predicates defining disjoint types; there is no such
procedure.Input-port? andoutput-port? should be included in this list of predicates.
An object may satisfy bothinput-port? andoutput-port?, but cannot satisfy any other
pair of predicates in this list.

